

### Example 6: Modeling Change over Time: Polynomial, Piecewise Slopes, and Exponential Trends

These example models come from Hoffman (2015) chapter 6. We will be examining change in response time (RT) in milliseconds over six practice sessions (balanced time) to a measure of processing speed in a sample of 101 older adults. Of interest first is the extent to which individual differences in change in RT can be described by polynomial models. Next we will examine alternative families of change (piecewise slopes and exponential models), followed by the use of log time as an approximation for a (truly nonlinear) exponential model.

#### SAS Syntax for Data Manipulation:

```
* Location for original SAS files for these models - change this path to your computer;
%LET filesave = C:\Dropbox\PilesOfVariance\Chapter6\SAS;
LIBNAME filesave "&filesave.";
* Defining macro variable for datafile name to be replaced in code below;
%LET datafile=work.Chapter6;

* Bringing data into work library and centering time for polynomial models;
DATA &datafile.; SET filesave.SAS_Chapter6;
    time1 = session - 1; LABEL time1 = "time1: Session (0=1)"; RUN;
```

#### Model 0: Saturated Means, Unstructured Variance Model (the answer key best baseline model)

```
TITLE1 "Model 0: Saturated Means, Unstructured Variance Model";
PROC MIXED DATA=&datafile. COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
    CLASS PersonID session;
    MODEL rt = session / SOLUTION CHISQ DDFM=Satterthwaite;
    REPEATED session / R RCORR TYPE=UN SUBJECT=PersonID;
    LSMEANS session /;
RUN; TITLE1;
```

The variable of *session* will be our categorical indicator for time—it will structure the **R** matrix via the REPEATED and CLASS statements given that we have balanced data. Here, *session* also appears on the MODEL statement so that we can estimate all possible mean differences across sessions.

Iteration History				
Iteration	Evaluations	-2 Res Log Like	Criterion	
0	1	9155.43252939		
1	1	<b>8229.78846855</b>	0.00000000	

Use this **-2LL** for your online homework (which provides three digits after the decimal).

Estimated R Matrix for PersonID 101						
Row	Col1	Col2	Col3	Col4	Col5	Col6
1	<b>301985</b>	235659	217994	202607	192154	195360
2	235659	<b>259150</b>	230217	213232	202092	193268
3	217994	230217	<b>233368</b>	205209	196919	188604
4	202607	213232	205209	<b>217544</b>	193676	185321
5	192154	202092	196919	193676	<b>212098</b>	187840
6	195360	193268	188604	185321	187840	<b>196733</b>

This **R matrix** UN structure lets all variances and covariances be what they want.  
  
THIS IS THE PATTERN we are trying to duplicate with our model for the variance.

Estimated R Correlation Matrix for PersonID 101						
Row	Col1	Col2	Col3	Col4	Col5	Col6
1	1.0000	0.8424	0.8212	0.7905	0.7593	0.8015
2	0.8424	1.0000	0.9361	0.8981	0.8620	0.8559
3	0.8212	0.9361	1.0000	0.9108	0.8851	0.8802
4	0.7905	0.8981	0.9108	1.0000	0.9016	0.8958
5	0.7593	0.8620	0.8851	0.9016	1.0000	0.9196
6	0.8015	0.8559	0.8802	0.8958	0.9196	1.0000

In REML, only parameters in the model for the variance count. -2LL comparisons in REML must have the same fixed effects in the model for the means.

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8229.8	<b>21</b>	8271.8	8273.4	8294.0	8326.7	8347.7

Solution for Fixed Effects

Effect	Session (1-6)	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept		1672.14	44.1345	100	37.89	<.0001
Session	1	289.76	32.7000	100	8.86	<.0001
Session	2	143.04	26.2031	100	5.46	<.0001
Session	3	77.8986	22.8842	100	3.40	0.0010
Session	4	45.6604	20.7853	100	2.20	0.0303
Session	5	35.0397	18.1168	100	1.93	0.0559
Session	6	0	.	.	.	.

What do the effects of sessions 1-5 tell us?

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
Session	5	100	16.72	<.0001

What does this F-test tell us?

Least Squares Means

Effect	Session (1-6)	Estimate	Standard Error	DF	t Value	Pr >  t
Session	1	1961.89	54.6805	100	35.88	<.0001
Session	2	1815.17	50.6541	100	35.83	<.0001
Session	3	1750.03	48.0684	100	36.41	<.0001
Session	4	1717.80	46.4101	100	37.01	<.0001
Session	5	1707.18	45.8255	100	37.25	<.0001
Session	6	1672.14	44.1345	100	37.89	<.0001

The **saturated means model** lets all session means be what they want.

**THIS IS THE PATTERN** we are trying to duplicate with our model for the means.

By what other name do you know this answer key model (i.e., when estimated in least squares)?

By what other name do you know this answer key model (i.e., when estimated in SEM)?

### Model 1b: Empty Means, Random Intercept Model (the worst baseline model)

```
TITLE1 "Model 1b: Empty Means, Random Intercept Model";
PROC MIXED DATA=&datafile. COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
  CLASS PersonID session;
  MODEL rt = / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT / G V VCORR TYPE=UN SUBJECT=PersonID;
  REPEATED session / R TYPE=VC SUBJECT=PersonID;
  ODS OUTPUT CovParms=CovEmpty; RUN; TITLE1;
```

Level 1:  $y_{ti} = \beta_{0i} + e_{ti}$   
 Level 2: Intercept:  $\beta_{0i} = \gamma_{00} + U_{0i}$

Iteration History

Iteration	Evaluations	-2 Res Log Like	Criterion
0	1	9228.60509392	
1	1	<b>8536.86086232</b>	0.00000000

Estimated R Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	44900					
2		44900				
3			44900			
4				44900		
5					44900	
6						44900

This **R matrix** VC structure (equal variance over time, no covariance of any kind) will be used repeatedly as we add fixed and random effects of time to the model.

Estimated G Matrix

Row	Effect	PersonID	Col1
1	Intercept	101	200883

Estimated V Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	<b>245783</b>	200883	200883	200883	200883	200883
2	200883	<b>245783</b>	200883	200883	200883	200883
3	200883	200883	<b>245783</b>	200883	200883	200883
4	200883	200883	200883	<b>245783</b>	200883	200883
5	200883	200883	200883	200883	<b>245783</b>	200883
6	200883	200883	200883	200883	200883	<b>245783</b>

This random intercept model predicts a compound symmetry pattern for the **V** matrix (equal variance, equal covariance over time).

Estimated V Correlation Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	1.0000	0.8173	0.8173	0.8173	0.8173	0.8173
2	0.8173	1.0000	0.8173	0.8173	0.8173	0.8173
3	0.8173	0.8173	1.0000	0.8173	0.8173	0.8173
4	0.8173	0.8173	0.8173	1.0000	0.8173	0.8173
5	0.8173	0.8173	0.8173	0.8173	1.0000	0.8173
6	0.8173	0.8173	0.8173	0.8173	0.8173	1.0000

This translates into equal correlation over time...

This correlation gets a special name of...?

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z	Pr > Z
UN(1,1)	PersonID	200883	29471	6.82	<.0001
Session	PersonID	44900	2825.63	15.89	<.0001

Null Model Likelihood Ratio Test		
DF	Chi-Square	Pr > ChiSq
1	691.74	<.0001

This is the test of whether we need the random intercept variance: it compares our random intercept model to the nested e-only model. Thus, it is a test of whether the ICC is significantly > 0.

Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8536.9	2	8540.9	8540.9	8543.0	8546.1	8548.1

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	1770.70	45.4206	100	38.98	<.0001

The empty means part of this model predicts no change on average (RT across six sessions is constant at 1770).

### Model 2a: Fixed Linear Time, Random Intercept Model

```
TITLE1 "Model 2a: Fixed Linear Time, Random Intercept Model";
PROC MIXED DATA=&datafile. COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
  CLASS PersonID session;
  MODEL rt = time1 / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT / G V VCORR TYPE=UN SUBJECT=PersonID;
  REPEATED session / R TYPE=VC SUBJECT=PersonID;
  * CovParms will be used for pseudo-R2,
  InfoCrit for LRT against next model;
  ODS OUTPUT CovParms=CovFixLin InfoCrit=FitFixLin;

  * Predicting means from model for each session;
  ESTIMATE "Intercept at Session=1 Time=0" intercept 1 time1 0;
  ESTIMATE "Intercept at Session=2 Time=1" intercept 1 time1 1;
  ESTIMATE "Intercept at Session=3 Time=2" intercept 1 time1 2;
  ESTIMATE "Intercept at Session=4 Time=3" intercept 1 time1 3;
  ESTIMATE "Intercept at Session=5 Time=4" intercept 1 time1 4;
  ESTIMATE "Intercept at Session=6 Time=5" intercept 1 time1 5;
RUN; TITLE1;
```

$$\text{Level 1: } y_{ti} = \beta_{0i} + \beta_{1i}(\text{Time}_{ti}) + e_{ti}$$

$$\text{Level 2: Intercept: } \beta_{0i} = \gamma_{00} + U_{0i}$$

$$\text{Linear Time: } \beta_{1i} = \gamma_{10}$$

Iteration History

Iteration	Evaluations	-2 Res Log Like	Criterion
0	1	9202.30103061	
1	1	<b>8414.68804514</b>	0.00000000

Estimated R Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	35662					
2		35662				
3			35662			
4				35662		
5					35662	
6						35662

Estimated G Matrix

Row	Effect	PersonID	Col1
1	Intercept	101	202422

Estimated V Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	238084	202422	202422	202422	202422	202422
2	202422	238084	202422	202422	202422	202422
3	202422	202422	238084	202422	202422	202422
4	202422	202422	202422	238084	202422	202422
5	202422	202422	202422	202422	238084	202422
6	202422	202422	202422	202422	202422	238084

The predicted V matrix still has a compound symmetry pattern because we have not yet added to the model for the variance (still a random intercept variance only in G).

Estimated V Correlation Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	1.0000	0.8502	0.8502	0.8502	0.8502	0.8502
2	0.8502	1.0000	0.8502	0.8502	0.8502	0.8502
3	0.8502	0.8502	1.0000	0.8502	0.8502	0.8502
4	0.8502	0.8502	0.8502	1.0000	0.8502	0.8502
5	0.8502	0.8502	0.8502	0.8502	1.0000	0.8502
6	0.8502	0.8502	0.8502	0.8502	0.8502	1.0000

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z
UN(1,1)	PersonID	202422	29470	6.87	<.0001
Session	PersonID	35662	2246.48	15.87	<.0001

Null Model Likelihood Ratio Test

DF	Chi-Square	Pr > ChiSq
1	787.61	<.0001

Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8414.7	2	8418.7	8418.7	8420.8	8423.9	8425.9

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	1899.63	46.7882	113	40.60	<.0001
time1	-51.5719	4.4918	504	-11.48	<.0001

Is the fixed linear time slope significant?  
How do we know?

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept at Session=1 Time=0	1899.63	46.7882	113	40.60	<.0001
Intercept at Session=2 Time=1	1848.06	45.9176	104	40.25	<.0001
Intercept at Session=3 Time=2	1796.49	45.4761	100	39.50	<.0001
Intercept at Session=4 Time=3	1744.92	45.4761	100	38.37	<.0001
Intercept at Session=5 Time=4	1693.34	45.9176	104	36.88	<.0001
Intercept at Session=6 Time=5	1641.77	46.7882	113	35.09	<.0001

What kind of change does this model predict?

What kind of individual differences in change does this model predict?

```
* Call macro to calculate pseudo R2;
%PseudoR2(NCov=2, CovFewer=CovEmpty, CovMore=CovFixLin);
```

**PseudoR2 (% Reduction) for CovEmpty vs. CovFixLin**

Name	CovParm	Subject	Estimate	StdErr	ZValue	ProbZ	PseudoR2
CovEmpty	UN(1,1)	PersonID	200883	29471	6.82	<.0001	.
CovEmpty	session	PersonID	44900	2825.63	15.89	<.0001	.
CovFixLin	UN(1,1)	PersonID	202422	29470	6.87	<.0001	-0.00766
CovFixLin	session	PersonID	35662	2246.48	15.87	<.0001	0.20575

Which variance did the fixed linear time slope explain?

Why did the other variance increase instead (i.e., have a negative pseudo-R<sup>2</sup> value)?

**Model 2b: Random Linear Time Model**

Level 1:  $y_{it} = \beta_{0i} + \beta_{1i}(\text{Time}_{it}) + e_{it}$   
 Level 2: Intercept:  $\beta_{0i} = \gamma_{00} + U_{0i}$   
 Linear Time:  $\beta_{1i} = \gamma_{10} + U_{1i}$

```
TITLE1 "Model 2b: Random Linear Time Model";
PROC MIXED DATA=&datafile. COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
  CLASS PersonID session;
  MODEL rt = time1 / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT time1 / G GCORR V VCORR TYPE=UN SUBJECT=PersonID;
  REPEATED session / R TYPE=VC SUBJECT=PersonID;
* CovParms will be used for pseudo-R2, InfoCrit for LRT against next model;
  ODS OUTPUT CovParms=CovRandLin InfoCrit=FitRandLin; RUN; TITLE1;
```

Iteration History

Iteration	Evaluations	-2 Res Log Like	Criterion
0	1	9202.30103061	
1	1	<b>8372.10246085</b>	0.00000000

Estimated R Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	27905					
2		27905				
3			27905			
4				27905		
5					27905	
6						27905

Estimated G Matrix

Row	Effect	PersonID	Col1	Col2
1	Intercept	101	253258	-12701
2	time1	101	-12701	2233.83

Estimated G Correlation Matrix

Row	Effect	PersonID	Col1	Col2
1	Intercept	101	1.0000	-0.5340
2	time1	101	-0.5340	1.0000

Estimated V Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
-----	------	------	------	------	------	------

1	<b>281163</b>	240557	227856	215155	202455	189754
2	240557	<b>257995</b>	219623	209156	198689	188222
3	227856	219623	<b>239295</b>	203157	194924	186691
4	215155	209156	203157	<b>225063</b>	191158	185159
5	202455	198689	194924	191158	<b>215298</b>	183627
6	189754	188222	186691	185159	183627	<b>210001</b>

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr >  Z
UN(1,1)	PersonID	253258	37897	6.68	<.0001
UN(2,1)	PersonID	-12701	3621.98	-3.51	0.0005
UN(2,2)	PersonID	2233.83	552.92	4.04	<.0001
Session	PersonID	27905	1963.42	14.21	<.0001

Null Model Likelihood Ratio Test

DF	Chi-Square	Pr > ChiSq
3	830.20	<.0001

Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8372.1	4	8380.1	8380.2	8384.3	8390.6	8394.6

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	1899.63	51.4998	100	36.89	<.0001
time1	-51.5719	6.1567	100	-8.38	<.0001

Is the random linear time slope significant?  
How do we know?

Do we need to compute pseudo-R<sup>2</sup> for this random linear time slope model relative to the previous fixed linear time slope, random intercept model?

```
* Call macro to calculate LRT for nested models;
%FitTest(FitFewer=FitFixLin, FitMore=FitRandLin);
```

Likelihood Ratio Test for FitFixLin vs. FitRandLin

Name	Neg2Log Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitFixLin	8414.7	2	8418.7	8423.9	.	.	.
FitRandLin	8372.1	4	8380.1	8390.6	42.5856	2	5.6579E-10

95% Random Effects Confidence Intervals that describe the predicted range of individual random effects:

$$\text{Random Effect 95\% CI} = \text{fixed effect} \pm (1.96 * \sqrt{\text{Random Variance}})$$

$$\text{Intercept 95\% CI} = \gamma_{00} \pm (1.96 * \sqrt{\tau_{U_0}^2}) \rightarrow 1,899.63 \pm (1.96 * \sqrt{253,258}) = 913 \text{ to } 2,886$$

$$\text{Linear Time Slope 95\% CI} = \gamma_{10} \pm (1.96 * \sqrt{\tau_{U_1}^2}) \rightarrow -51.57 \pm (1.96 * \sqrt{2,233.83}) = -145 \text{ to } 42$$

*Is it a problem that the CI for the linear time slope overlaps 0? What does this mean?*

**Begin Polynomial Models → Model 3a: Fixed Quadratic, Random Linear Time Model**

Level 1:	$y_{it} = \beta_{0i} + \beta_{1i}(\text{Time}_{it}) + \beta_{2i}(\text{Time}_{it})^2 + e_{it}$
Level 2: Intercept:	$\beta_{0i} = \gamma_{00} + U_{0i}$
Linear Time:	$\beta_{1i} = \gamma_{10} + U_{1i}$
Quadratic Time:	$\beta_{2i} = \gamma_{20}$

```

TITLE1 "Model 3a: Fixed Quadratic, Random Linear Time Model";
PROC MIXED DATA=&datafile. COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
CLASS PersonID session;
MODEL rt = time1 time1*time1 / SOLUTION DDFM=Satterthwaite;
RANDOM INTERCEPT time1 / G GCORR V VCORR TYPE=UN SUBJECT=PersonID;
REPEATED session / R TYPE=VC SUBJECT=PersonID;
ODS OUTPUT CovParms=CovFixQuad InfoCrit=FitFixQuad;
* Predicting means from model for each session;
ESTIMATE "Intercept at Session=1 Time=0" intercept 1 time1 0 time1*time1 0;
ESTIMATE "Intercept at Session=2 Time=1" intercept 1 time1 1 time1*time1 1;
ESTIMATE "Intercept at Session=3 Time=2" intercept 1 time1 2 time1*time1 4;
ESTIMATE "Intercept at Session=4 Time=3" intercept 1 time1 3 time1*time1 9;
ESTIMATE "Intercept at Session=5 Time=4" intercept 1 time1 4 time1*time1 16;
ESTIMATE "Intercept at Session=6 Time=5" intercept 1 time1 5 time1*time1 25;
* Predicting linear rate of change at each session (linear changes by 2*quad);
ESTIMATE "Linear Slope at Session=1 Time=0" time1 1 time1*time1 0;
ESTIMATE "Linear Slope at Session=2 Time=1" time1 1 time1*time1 2;
ESTIMATE "Linear Slope at Session=3 Time=2" time1 1 time1*time1 4;
ESTIMATE "Linear Slope at Session=4 Time=3" time1 1 time1*time1 6;
ESTIMATE "Linear Slope at Session=5 Time=4" time1 1 time1*time1 8;
ESTIMATE "Linear Slope at Session=6 Time=5" time1 1 time1*time1 10;
RUN; TITLE1;

```

Because twice the quadratic slope is how the linear slope changes per unit time, the value for time1 used in estimating the linear slope per session gets multiplied by 2.

Iteration History

Iteration	Evaluations	-2 Res Log Like	Criterion
0	1	9193.25780414	
1	1	<b>8341.47727191</b>	0.00000000

Estimated R Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	26176					
2		26176				
3			26176			
4				26176		
5					26176	
6						26176

Estimated G Matrix

Row	Effect	PersonID	Col1	Col2
1	Intercept	101	254164	-12948
2	time1	101	-12948	2332.67

Estimated G Correlation Matrix

Row	Effect	PersonID	Col1	Col2
1	Intercept	101	1.0000	-0.5318
2	time1	101	-0.5318	1.0000

Estimated V Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	<b>280339</b>	241216	228268	215320	202372	189424
2	241216	<b>256776</b>	219985	209370	198755	188140
3	228268	219985	<b>237879</b>	203420	195138	186855
4	215320	209370	203420	<b>223646</b>	191521	185571
5	202372	198755	195138	191521	<b>214079</b>	184286
6	189424	188140	186855	185571	184286	<b>209178</b>

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z	Pr >  Z
UN(1,1)	PersonID	254164	37896	6.71	<.0001
UN(2,1)	PersonID	-12948	3620.70	-3.58	0.0003
UN(2,2)	PersonID	2332.67	551.58	4.23	<.0001
Session	PersonID	26176	1844.01	14.20	<.0001

Null Model Likelihood Ratio Test

DF	Chi-Square	Pr > ChiSq
3	851.78	<.0001

Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8341.5	4	8349.5	8349.5	8353.7	8359.9	8363.9

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	1945.85	52.2433	106	37.25	<.0001
time1	-120.90	14.5415	502	-8.31	<.0001
time1*time1	13.8656	2.6348	403	5.26	<.0001

Is the fixed quadratic time slope significant? How do we know?

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept at Session=1 Time=0	1945.85	52.2433	106	37.25	<.0001
Intercept at Session=2 Time=1	1838.82	48.6084	100	37.83	<.0001
Intercept at Session=3 Time=2	1759.51	46.8223	105	37.58	<.0001
Intercept at Session=4 Time=3	1707.94	45.2925	105	37.71	<.0001
Intercept at Session=5 Time=4	1684.10	44.0458	100	38.24	<.0001
Intercept at Session=6 Time=5	1687.99	44.9976	108	37.51	<.0001
Linear Slope at Session=1 Time=0	-120.90	14.5415	502	-8.31	<.0001
Linear Slope at Session=2 Time=1	-93.1687	10.0191	419	-9.30	<.0001
Linear Slope at Session=3 Time=2	-65.4375	6.6968	139	-9.77	<.0001
Linear Slope at Session=4 Time=3	-37.7062	6.6968	139	-5.63	<.0001
Linear Slope at Session=5 Time=4	-9.9750	10.0191	419	-1.00	0.3200
Linear Slope at Session=6 Time=5	17.7562	14.5415	502	1.22	0.2226

What happens to the linear time slope over sessions because of the quadratic time slope (i.e., which of the four kinds of interaction is this quadratic effect of time)?

Why did the fixed linear time slope change from the previous model (-52 versus -120)?

```
* Call macro to calculate pseudo R2;
%PseudoR2(NCov=4, CovFewer=CovRandLin, CovMore=CovFixQuad);
```

PseudoR2 (% Reduction) for CovRandLin vs. CovFixQuad

Name	CovParm	Subject	Estimate	StdErr	ZValue	ProbZ	PseudoR2
CovRandLin	UN(1,1)	PersonID	253258	37897	6.68	<.0001	.
CovRandLin	UN(2,2)	PersonID	2233.83	552.92	4.04	<.0001	.
CovRandLin	session	PersonID	27905	1963.42	14.21	<.0001	.
CovFixQuad	UN(1,1)	PersonID	254164	37896	6.71	<.0001	-0.003577
CovFixQuad	UN(2,2)	PersonID	2332.67	551.58	4.23	<.0001	-0.044244
CovFixQuad	session	PersonID	26176	1844.01	14.20	<.0001	0.061980

Which variance did the fixed quadratic time slope explain?

Why did the other variances increase instead (i.e., have negative pseudo-R<sup>2</sup> values)?



### Model 3b: Random Quadratic Time Model

Level 1:  $y_{ti} = \beta_{0i} + \beta_{1i}(\text{Time}_{ti}) + \beta_{2i}(\text{Time}_{ti})^2 + e_{ti}$

Level 2: Intercept:  $\beta_{0i} = \gamma_{00} + U_{0i}$

Linear Time:  $\beta_{1i} = \gamma_{10} + U_{1i}$

Quadratic Time:  $\beta_{2i} = \gamma_{20} + U_{2i}$

```
TITLE1 "Model 3b: Random Quadratic Model";
PROC MIXED DATA=&datafile. COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
  CLASS PersonID session;
  MODEL rt = time1 time1*time1 / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT time1 time1*time1 / G GCORR V VCORR TYPE=UN SUBJECT=PersonID;
  REPEATED session / R TYPE=VC SUBJECT=PersonID;
  ODS OUTPUT InfoCrit=FitRandQuad; RUN; TITLE1;
```

Iteration History

Iteration	Evaluations	-2 Res Log Like	Criterion
0	1	9193.25780414	
1	1	<b>8302.74566856</b>	0.00000000

Estimated R Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	20298					
2		20298				
3			20298			
4				20298		
5					20298	
6						20298

Estimated G Matrix

Row	Effect	PersonID	Col1	Col2	Col3
1	Intercept	101	276206	-35734	3901.96
2	time1	101	-35734	25840	-3903.32
3	time1*time1	101	3901.96	-3903.32	634.47

Estimated G Correlation Matrix

Row	Effect	PersonID	Col1	Col2	Col3
1	Intercept	101	1.0000	-0.4230	0.2948
2	time1	101	-0.4230	1.0000	-0.9640
3	time1*time1	101	0.2948	-0.9640	1.0000

Estimated V Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	<b>296504</b>	244374	220346	204122	195702	195085
2	244374	<b>251508</b>	219312	208680	199315	191215
3	220346	219312	<b>235842</b>	209043	199808	187840
4	204122	208680	209043	<b>225508</b>	197182	184958
5	195702	199315	199808	197182	<b>211735</b>	182571
6	195085	191215	187840	184958	182571	<b>200977</b>

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr >  Z
UN(1,1)	PersonID	276206	41442	6.66	<.0001
UN(2,1)	PersonID	-35734	11941	-2.99	0.0028
UN(2,2)	PersonID	25840	5864.41	4.41	<.0001
UN(3,1)	PersonID	3901.96	1949.06	2.00	0.0453
UN(3,2)	PersonID	-3903.32	982.61	-3.97	<.0001
UN(3,3)	PersonID	634.47	172.37	3.68	0.0001
Session	PersonID	20298	1649.11	12.31	<.0001

Null Model Likelihood Ratio Test

DF	Chi-Square	Pr > ChiSq
6	890.51	<.0001

Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8302.7	7	8316.7	8316.9	8324.2	8335.1	8342.1

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	1945.85	53.8497	100	36.13	<.0001
time1	-120.90	20.0476	100	-6.03	<.0001
time1*time1	13.8656	3.4154	100	4.06	<.0001

Is the random quadratic time slope significant? How do we know?

Do we need to compute pseudo-R2 for this random quadratic time slope model relative to the previous fixed quadratic time slope, random linear time slope model?

```
* Call macro to calculate LRT for nested models;
%FitTest(FitFewer=FitFixQuad, FitMore=FitRandQuad);
```

**Likelihood Ratio Test for FitFixQuad vs. FitRandQuad**

Name	Neg2Log Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitFixQuad	8341.5	4	8349.5	8359.9	.	.	.
FitRandQuad	8302.7	7	8316.7	8335.1	38.7316	3	1.9784E-8

95% Random Effect Confidence Intervals that describe the *predicted* range of individual random effects:

$$\text{Random Effect 95\% CI} = \text{fixed effect} \pm (1.96 * \sqrt{\text{Random Variance}})$$

$$\text{Intercept 95\% CI} = \gamma_{00} \pm (1.96 * \sqrt{\tau_{U_0}^2}) \rightarrow 1,945.9 \pm (1.96 * \sqrt{276,209}) = 916 \text{ to } 2,976$$

$$\text{Linear Time Slope 95\% CI} = \gamma_{10} \pm (1.96 * \sqrt{\tau_{U_1}^2}) \rightarrow -120.9 \pm (1.96 * \sqrt{25,840}) = -436 \text{ to } 194$$

$$\text{Quadratic Time Slope 95\% CI} = \gamma_{20} \pm (1.96 * \sqrt{\tau_{U_2}^2}) \rightarrow 13.9 \pm (1.96 * \sqrt{634}) = -36 \text{ to } 63$$

*Is it a problem that the CIs for the linear and quadratic time slopes overlap 0? What does this mean?*

**Bonus question: is a VC R matrix (equal residual variance, no residual covariance) sufficient?**

```
TITLE1 "Testing AR1 Residual Correlation Across Time";
PROC MIXED DATA=&datafile. COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
CLASS PersonID session;
MODEL rt = time1 time1*time1 / SOLUTION DDFM=Satterthwaite;
RANDOM INTERCEPT time1 time1*time1 / G V VCORR TYPE=UN SUBJECT=PersonID;
REPEATED session / R RCORR TYPE=AR(1) SUBJECT=PersonID;
ODS OUTPUT InfoCrit=FitAR1; RUN; TITLE1;
```

Estimated R Correlation Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	1.0000	0.1711	0.02927	0.005008	0.000857	0.000147
2	0.1711	1.0000	0.1711	0.02927	0.005008	0.000857
3	0.02927	0.1711	1.0000	0.1711	0.02927	0.005008
4	0.005008	0.02927	0.1711	1.0000	0.1711	0.02927
5	0.000857	0.005008	0.02927	0.1711	1.0000	0.1711
6	0.000147	0.000857	0.005008	0.02927	0.1711	1.0000

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr >  Z
UN(1,1)	PersonID	272126	41822	6.51	<.0001
UN(2,1)	PersonID	-35004	12302	-2.85	0.0044
UN(2,2)	PersonID	24064	6386.03	3.77	<.0001
UN(3,1)	PersonID	3941.63	1996.01	1.97	0.0483
UN(3,2)	PersonID	-3554.21	1076.64	-3.30	0.0010
UN(3,3)	PersonID	547.63	194.97	2.81	0.0025
<b>AR(1)</b>	<b>PersonID</b>	<b>0.1711</b>	<b>0.1286</b>	<b>1.33</b>	<b>0.1833</b>
Residual		24534	4585.43	5.35	<.0001

Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8300.7	8	8316.7	8316.9	8325.1	8337.6	8345.6

\* Call macro to calculate LRT for nested models;  
 %FitTest(FitFewer=FitRandQuad, FitMore=FitAR1);

Likelihood Ratio Test for FitRandQuad vs. FitAR1

Name	Neg2Log Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitRandQuad	8302.7	7	8316.7	8335.1	.	.	.
FitAR1	8300.7	8	8316.7	8337.6	2.07551	1	0.14968

```
TITLE1 "Testing Lag-1 Toeplitz Residual Covariance Across Time";
PROC MIXED DATA=&datafile. COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
  CLASS PersonID session;
  MODEL rt = time1 time1*time1 / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT time1 time1*time1 / G V VCORR TYPE=UN SUBJECT=PersonID;
  REPEATED session / R RCORR TYPE=TOEP(2) SUBJECT=PersonID;
  ODS OUTPUT InfoCrit=FitTOEP2; RUN; TITLE1;
```

Estimated R Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	23059	2856.63				
2	2856.63	23059	2856.63			
3		2856.63	23059	2856.63		
4			2856.63	23059	2856.63	
5				2856.63	23059	2856.63
6					2856.63	23059

Estimated R Correlation Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	1.0000	0.1239				
2	0.1239	1.0000	0.1239			
3		0.1239	1.0000	0.1239		
4			0.1239	1.0000	0.1239	
5				0.1239	1.0000	0.1239
6					0.1239	1.0000

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr >  Z
UN(1,1)	PersonID	273602	41647	6.57	<.0001
UN(2,1)	PersonID	-35273	12222	-2.89	0.0039
UN(2,2)	PersonID	24667	6180.41	3.99	<.0001
UN(3,1)	PersonID	3929.32	1987.62	1.98	0.0481
UN(3,2)	PersonID	-3664.04	1038.33	-3.53	0.0004
UN(3,3)	PersonID	573.36	184.45	3.11	0.0009
<b>TOEP(2)</b>	<b>PersonID</b>	<b>2856.63</b>	<b>2130.45</b>	<b>1.34</b>	<b>0.1800</b>
Residual		23059	2867.62	8.04	<.0001



Neg2LogLike	Parms	Information Criteria				CAIC
		AIC	AICC	HQIC	BIC	
8300.9	8	8316.9	8317.1	8325.4	8337.8	8345.8

```
* Call macro to calculate LRT for nested models;  
%FitTest(FitFewer=FitRandQuad, FitMore=FitTOEP2);
```

Likelihood Ratio Test for FitRandQuad vs. FitTOEP2

Name	Neg2Log Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitRandQuad	8302.7	7	8316.7	8335.1	.	.	.
FitTOEP2	8300.9	8	8316.9	8337.8	1.83947	1	0.17501

Given that neither R matrix improved model fit, I'd say this is as good as it gets for quadratic time. So how did we do? Let's compare model predictions in terms of means (top) and variances (bottom)?



### Bonus Material: Testing Absolute Fit when using REML

As shown as Model 0, the saturated means, unstructured variance model is the best-fitting model for each side (means and variances). However, when using REML, we cannot do a model comparison against our random quadratic model, because the models cannot differ in their fixed effects for the  $-2LL$  (LRT) to be valid. Instead, we can test the absolute fit for each side of the model separately.

```
TITLE1 "Model 3b: Random Quadratic Model (Best Model So Far, Repeated)";
PROC MIXED DATA=&datafile. COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
  CLASS PersonID session;
  MODEL rt = time1 time1*time1 / SOLUTION CL CHISQ DDFM=Satterthwaite;
  RANDOM INTERCEPT time1 time1*time1 / G GCORR V VCORR TYPE=UN SUBJECT=PersonID;
  REPEATED session / R RCORR TYPE=VC SUBJECT=PersonID;
  ODS OUTPUT InfoCrit=FitRandQuad; RUN; TITLE1;
```

The means side can be tested using the *same random quadratic time slopes* in each version:

```
TITLE1 "Testing Absolute Fit of the Means Model (Using Random Quadratic Variance Model)";
PROC MIXED DATA=&datafile. COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
  CLASS PersonID session;
  * Add session to the model to fully saturate the means model;
  MODEL rt = time1 time1*time1 session / SOLUTION CL CHISQ DDFM=Satterthwaite HTYPE=1;
  RANDOM INTERCEPT time1 time1*time1 / G GCORR V VCORR TYPE=UN SUBJECT=PersonID;
  REPEATED session / R RCORR TYPE=VC SUBJECT=PersonID; RUN; TITLE1;
```

Solution for Fixed Effects						
Effect	Session	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept		1603.14	271.78	323	5.90	<.0001
time1		74.8457	138.82	312	0.54	0.5902
time1*time1		-12.2095	17.3760	323	-0.70	0.4828
session	1	358.75	267.06	300	1.34	0.1802
session	2	149.39	147.22	300	1.01	0.3110
session	3	46.0365	62.7733	300	0.73	0.4639
session	4	0	.	.	.	.
session	5	0	.	.	.	.
session	6	0	.	.	.	.

Type 1 Tests of Fixed Effects						
Effect	Num DF	Den DF	Chi-Square	F Value	Pr > ChiSq	Pr > F
time1	1	100	70.17	70.17	<.0001	<.0001
time1*time1	1	100	16.48	16.48	<.0001	<.0001
session	3	300	9.07	3.02	0.0284	<b>0.0299</b>

There are now 6 fixed effects for the 6 means, such that this is now equivalent to a saturated means model (even if the coefficients are largely uninterpretable).

SAS figures out how many extra contrasts are needed to saturate and only estimates that many.

Htype=1 is a sequential test in order of entry. The multivariate Wald test for session indicates that the 3 extra session contrasts improved model fit (which is bad news).

The variance side can be tested using the *same fixed time slopes* in each version:

```
TITLE1 "Testing Absolute Fit of the Variance Model (Using Fixed Quadratic Means Model)";
PROC MIXED DATA=&datafile. COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
  CLASS PersonID session;
  MODEL rt = time1 time1*time1 / SOLUTION CL CHISQ DDFM=Satterthwaite;
  REPEATED session / R RCORR TYPE=UN SUBJECT=PersonID;
  ODS OUTPUT InfoCrit=FitRandUN; RUN; TITLE1;
  * Call macro to calculate LRT for nested models;
  %FitTest(FitFewer=FitRandQuad, FitMore=FitRandUN);
```

(Other irrelevant output omitted)

#### Likelihood Ratio Test for FitRandQuad vs. FitRandUN

Name	Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitRandQuad	8302.7	7	8316.7	8335.1	.	.	.
FitRandUN	8267.0	21	8309.0	8363.9	35.7580	14	<b>.001134108</b>

What does the significant  $p$ -value for this model comparison indicate?

## Piecewise Slopes Models

### SAS Syntax for Data Manipulation:

```
* Example for BALANCED DATA (everyone on same measurement schedule);
* Creating pieces with intercept at session 1, breakpoint at session 2;
* Making predictors for 2 Direct Slopes and Slope+Deviation Slope models;
DATA &datafile.; SET &datafile.;
time1 = session - 1; LABEL time1 = "time1: Session (0=1)";
  IF session = 1 THEN DO; Slope12 = 0; Slope26 = 0; END;
ELSE IF session = 2 THEN DO; Slope12 = 1; Slope26 = 0; END;
ELSE IF session = 3 THEN DO; Slope12 = 1; Slope26 = 1; END;
ELSE IF session = 4 THEN DO; Slope12 = 1; Slope26 = 2; END;
ELSE IF session = 5 THEN DO; Slope12 = 1; Slope26 = 3; END;
ELSE IF session = 6 THEN DO; Slope12 = 1; Slope26 = 4; END;
LABEL Slope12 = "Slope12: Early Practice Slope (Session 1-2)"
      Slope26 = "Slope26: Later Practice Slope (Session 2-6)"; RUN;
```

### Model 4a: Fixed Slope12, Fixed Slope26, Random Intercept Model

Level 1:  $y_{ti} = \beta_{0i} + \beta_{1i}(\text{Slope12}_{ti}) + \beta_{2i}(\text{Slope26}_{ti}) + e_{ti}$

Level 2: Intercept:  $\beta_{0i} = \gamma_{00} + U_{0i}$

Slope12:  $\beta_{1i} = \gamma_{10}$

Slope26:  $\beta_{2i} = \gamma_{20}$

```
TITLE1 "Model 4a: Fixed Slope12, Fixed Slope26, Random Intercept Model";
PROC MIXED DATA=&datafile. COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
  CLASS PersonID session;
  MODEL rt = Slope12 Slope26 / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT / G V VCORR TYPE=UN SUBJECT=PersonID;
  REPEATED session / R TYPE=VC SUBJECT=PersonID;
  ODS OUTPUT CovParms=CovFix12Fix26 InfoCrit=FitFix12Fix26;
  ESTIMATE "Intercept at Session=1 Time=0" Intercept 1 Slope12 0 Slope26 0;
  ESTIMATE "Intercept at Session=2 Time=1" Intercept 1 Slope12 1 Slope26 0;
  ESTIMATE "Intercept at Session=3 Time=2" Intercept 1 Slope12 1 Slope26 1;
  ESTIMATE "Intercept at Session=4 Time=3" Intercept 1 Slope12 1 Slope26 2;
  ESTIMATE "Intercept at Session=5 Time=4" Intercept 1 Slope12 1 Slope26 3;
  ESTIMATE "Intercept at Session=6 Time=5" Intercept 1 Slope12 1 Slope26 4;
  ESTIMATE "Difference between slope12 and slope26" Slope12 -1 Slope26 1; RUN; TITLE1;
```

Iteration History				
Iteration	Evaluations	-2 Res	Log Like	Criterion
0	1	9188.48345679		
1	1	<b>8382.68712287</b>		0.00000000

Use this **-2LL** for your online homework (which provides three digits after the decimal).

Estimated R Matrix for PersonID 101						
Row	Col1	Col2	Col3	Col4	Col5	Col6
1	34098					
2		34098				
3			34098			
4				34098		
5					34098	
6						34098

This **R matrix** VC structure (equal variance over time, no covariance of any kind) will be used repeatedly as we add fixed and random piecewise slopes to the model.

Estimated G Matrix			
Row	Effect	PersonID	Col1
1	Intercept	101	202683

Estimated V Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	<b>236781</b>	202683	202683	202683	202683	202683
2	202683	<b>236781</b>	202683	202683	202683	202683
3	202683	202683	<b>236781</b>	202683	202683	202683
4	202683	202683	202683	<b>236781</b>	202683	202683
5	202683	202683	202683	202683	<b>236781</b>	202683
6	202683	202683	202683	202683	202683	<b>236781</b>

This random intercept model predicts a compound symmetry pattern for the V matrix (equal variance, equal covariance over time).

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr >  Z
UN(1,1)	PersonID	202683	29470	6.88	<.0001
Session	PersonID	34098	2150.11	15.86	<.0001

Null Model Likelihood Ratio Test

DF	Chi-Square	Pr > ChiSq
1	805.80	<.0001

Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8382.7	2	8386.7	8386.7	8388.8	8391.9	8393.9

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	1961.89	48.4187	129	40.52	<.0001
Slope12	-163.64	23.2415	503	-7.04	<.0001
Slope26	-32.8932	5.8104	503	-5.66	<.0001


Are the fixed piecewise slopes significant? How do we know?

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept at Session=1 Time=0	1961.89	48.4187	129	40.52	<.0001
Intercept at Session=2 Time=1	1798.25	47.0035	115	38.26	<.0001
Intercept at Session=3 Time=2	1765.36	45.9134	104	38.45	<.0001
Intercept at Session=4 Time=3	1732.46	45.5443	101	38.04	<.0001
Intercept at Session=5 Time=4	1699.57	45.9134	104	37.02	<.0001
Intercept at Session=6 Time=5	1666.68	47.0035	115	35.46	<.0001
Difference between slope12 and slope26	130.75	26.6265	503	4.91	<.0001

Number-Match 3

Session	1	2	3	4	5	6
Early Practice → Slope12 =	0	1	1	1	1	1
Later Practice → Slope26 =	0	0	1	2	3	4



```
* Call macro to calculate pseudo R2;
%PseudoR2(NCov=2, CovFewer=CovEmpty, CovMore=CovFix12Fix26);
```

**PseudoR2 (% Reduction) for CovEmpty vs. CovFix12Fix26**

Name	CovParm	Subject	Estimate	StdErr	ZValue	ProbZ	PseudoR2
CovEmpty	UN(1,1)	PersonID	200883	29471	6.82	<.0001	.
CovEmpty	session	PersonID	44900	2825.63	15.89	<.0001	.
CovFix12Fix26	UN(1,1)	PersonID	202683	29470	6.88	<.0001	-0.00896
CovFix12Fix26	session	PersonID	34098	2150.11	15.86	<.0001	0.24058

**Model 4b: Random Slope12, Fixed Slope26 Model**

Level 1:  $y_{ti} = \beta_{0i} + \beta_{1i}(\text{Slope12}_{ti}) + \beta_{2i}(\text{Slope26}_{ti}) + e_{ti}$   
 Level 2: Intercept:  $\beta_{0i} = \gamma_{00} + U_{0i}$   
 Slope12:  $\beta_{1i} = \gamma_{10} + U_{1i}$   
 Slope26:  $\beta_{2i} = \gamma_{20}$

```
TITLE1 "Model 4b: Random Slope12, Fixed Slope26 Model";
PROC MIXED DATA=&datafile. COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
CLASS PersonID session;
MODEL rt = Slope12 Slope26 / SOLUTION DDFM=Satterthwaite;
RANDOM INTERCEPT Slope12 / G GCORR V VCORR TYPE=UN SUBJECT=PersonID;
REPEATED session / R TYPE=VC SUBJECT=PersonID;
ODS OUTPUT InfoCrit=FitRand12Fix26; RUN; TITLE1;
```

Iteration History

Iteration	Evaluations	-2 Res Log Like	Criterion
0	1	9188.48345679	
1	1	<b>8319.61326151</b>	0.00000000

Estimated G Matrix

Row	Effect	PersonID	Col1	Col2
1	Intercept	101	277818	-69063
2	Slope12	101	-69063	59941

Estimated G Correlation Matrix

Row	Effect	PersonID	Col1	Col2
1	Intercept	101	1.0000	-0.5352
2	Slope12	101	-0.5352	1.0000

Estimated V Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	<b>301985</b>	208755	208755	208755	208755	208755
2	208755	<b>223800</b>	199632	199632	199632	199632
3	208755	199632	<b>223800</b>	199632	199632	199632
4	208755	199632	199632	<b>223800</b>	199632	199632
5	208755	199632	199632	199632	<b>223800</b>	199632
6	208755	199632	199632	199632	199632	<b>223800</b>

How would we describe the pattern of variance and covariances in **V** now?

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z	Pr >  Z
UN(1,1)	PersonID	277818	42741	6.50	<.0001
UN(2,1)	PersonID	-69063	18932	-3.65	0.0003
UN(2,2)	PersonID	59941	12743	4.70	<.0001
Session	PersonID	24168	1702.53	14.20	<.0001



Null Model Likelihood Ratio Test

DF	Chi-Square	Pr > ChiSq
3	868.87	<.0001

Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8319.6	4	8327.6	8327.7	8331.8	8338.1	8342.1

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	1961.89	54.6805	100	35.88	<.0001
Slope12	-163.64	31.2462	123	-5.24	<.0001
Slope26	-32.8932	4.8916	403	-6.72	<.0001

Is the random slope12 significant?  
How do we know?

```
* Call macro to calculate LRT for nested models;
%FitTest(FitFewer=FitFix12Fix26, FitMore=FitRand12Fix26);
```

Likelihood Ratio Test for FitFix12Fix26 vs. FitRand12Fix26

Name	Neg2Log Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitFix12Fix26	8382.7	2	8386.7	8391.9	.	.	.
FitRand12Fix26	8319.6	4	8327.6	8338.1	63.0739	2	2.0095E-14

**Model 4c: Random Slope12, Random Slope26 Model**

$$\text{Level 1: } y_{ti} = \beta_{0i} + \beta_{1i}(\text{Slope12}_{ti}) + \beta_{2i}(\text{Slope26}_{ti}) + e_{ti}$$

$$\text{Level 2: Intercept: } \beta_{0i} = \gamma_{00} + U_{0i}$$

$$\text{Slope12: } \beta_{1i} = \gamma_{10} + U_{1i}$$

$$\text{Slope26: } \beta_{2i} = \gamma_{20} + U_{2i}$$

```
TITLE1 "Model 4c: Random Slope12, Random Slope26 Model";
PROC MIXED DATA=&datafile. COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
CLASS PersonID session;
MODEL rt = Slope12 Slope26 / SOLUTION DDFM=Satterthwaite;
RANDOM INTERCEPT Slope12 Slope26 / G GCORR V VCORR TYPE=UN SUBJECT=PersonID;
REPEATED session / R TYPE=VC SUBJECT=PersonID;
ODS OUTPUT InfoCrit=FitRand12Rand26; RUN; TITLE1;
```

Iteration History

Iteration	Evaluations	-2 Res Log Like	Criterion
0	1	9188.48345679	
1	1	<b>8275.37431715</b>	0.00000000

Estimated G Matrix

Row	Effect	PersonID	Col1	Col2	Col3
1	Intercept	101	284312	-54270	-10644
2	Slope12	101	-54270	63954	-1672.30
3	Slope26	101	-10644	-1672.30	2617.28

Estimated G Correlation Matrix

Row	Effect	PersonID	Col1	Col2	Col3
1	Intercept	101	1.0000	-0.4025	-0.3902
2	Slope12	101	-0.4025	1.0000	-0.1293
3	Slope26	101	-0.3902	-0.1293	1.0000

Estimated V Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	<b>301985</b>	230042	219399	208755	198111	187467
2	230042	<b>257400</b>	227410	215094	202778	190462
3	219399	227410	<b>235385</b>	208013	198314	188615
4	208755	215094	208013	<b>218604</b>	193850	186768
5	198111	202778	198314	193850	<b>207059</b>	184921
6	187467	190462	188615	186768	184921	<b>200747</b>

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr >  Z
UN(1,1)	PersonID	284312	42731	6.65	<.0001
UN(2,1)	PersonID	-54270	18230	-2.98	0.0029
UN(2,2)	PersonID	63954	13244	4.83	<.0001
UN(3,1)	PersonID	-10644	3791.26	-2.81	0.0050
UN(3,2)	PersonID	-1672.30	2097.03	-0.80	0.4252
UN(3,3)	PersonID	2617.28	636.48	4.11	<.0001
Session	PersonID	17673	1435.84	12.31	<.0001

Null Model Likelihood Ratio Test

DF	Chi-Square	Pr > ChiSq
6	913.11	<.0001

Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8275.4	7	8289.4	8289.6	8296.8	8307.7	8314.7

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	1961.89	54.6805	100	35.88	<.0001
Slope12	-163.64	30.2188	100	-5.42	<.0001
Slope26	-32.8932	6.5888	100	-4.99	<.0001

Is the random slope26 significant?  
How do we know?

```
* Call macro to calculate LRT for nested models;
%FitTest(FitFewer=FitRand12Fix26, FitMore=FitRand12Rand26);
```

Likelihood Ratio Test for FitRand12Fix26 vs. FitRand12Rand26

Name	Neg2Log Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitRand12Fix26	8319.6	4	8327.6	8338.1	.	.	.
FitRand12Rand26	8275.4	7	8289.4	8307.7	44.2389	3	1.3427E-9

95% Random Effect Confidence Intervals that describe the *predicted* range of individual random effects:

$$\text{Random Effect 95\% CI} = \text{fixed effect} \pm \left(1.96 * \sqrt{\text{Random Variance}}\right)$$

$$\text{Intercept 95\% CI} = \gamma_{00} \pm \left(1.96 * \sqrt{\tau_{U_0}^2}\right) \rightarrow 1,961.9 \pm \left(1.96 * \sqrt{284,312}\right) = 917 \text{ to } 3,007$$

$$\text{Slope12 95\% CI} = \gamma_{10} \pm \left(1.96 * \sqrt{\tau_{U_1}^2}\right) \rightarrow -163.6 \pm \left(1.96 * \sqrt{63,954}\right) = -659 \text{ to } 322$$

$$\text{Slope26 95\% CI} = \gamma_{20} \pm \left(1.96 * \sqrt{\tau_{U_2}^2}\right) \rightarrow -32.9 \pm \left(1.96 * \sqrt{2,617}\right) = -133 \text{ to } 67$$

So far we've examined one way to fit piecewise slopes models—direct slopes that represent the change during each time period. Let's now examine an alternative specification—**slope + deviation slope**, which can be useful in examining individual differences in differential change between time periods.

**Model 5a: Fixed Slope, Fixed Deviation Slope, Random Intercept Model (Equivalent to 4a)**

Level 1:  $y_{ti} = \beta_{0i} + \beta_{1i}(\text{Time}_{ti}) + \beta_{2i}(\text{Slope26}_{ti}) + e_{ti}$   
 Level 2: Intercept:  $\beta_{0i} = \gamma_{00} + U_{0i}$   
 Time:  $\beta_{1i} = \gamma_{10}$   
 Slope26:  $\beta_{2i} = \gamma_{20}$

Session	1	2	3	4	5	6
Time → Slope16 =	0	1	2	3	4	5
Deviation → Slope26 =	0	0	1	2	3	4


```
TITLE1 "Model 5a: Fixed Slope, Fixed Deviation Slope, Random Intercept Model";
PROC MIXED DATA=&datafile. COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
  CLASS PersonID session;
  MODEL rt = Time1 Slope26 / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT / G V VCORR TYPE=UN SUBJECT=PersonID;
  REPEATED session / R TYPE=VC SUBJECT=PersonID;
  ODS OUTPUT CovParms=CovFix16Fix26 InfoCrit=FitFix16Fix26;
  ESTIMATE "Intercept at Session=1 Time=0" Intercept 1 Time1 0 Slope26 0;
  ESTIMATE "Intercept at Session=2 Time=1" Intercept 1 Time1 1 Slope26 0;
  ESTIMATE "Intercept at Session=3 Time=2" Intercept 1 Time1 2 Slope26 1;
  ESTIMATE "Intercept at Session=4 Time=3" Intercept 1 Time1 3 Slope26 2;
  ESTIMATE "Intercept at Session=5 Time=4" Intercept 1 Time1 4 Slope26 3;
  ESTIMATE "Intercept at Session=6 Time=5" Intercept 1 Time1 5 Slope26 4;
  ESTIMATE "Rate of change from session 2-6" Time1 1 Slope26 1; RUN; TITLE1;
```

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	1961.89	48.4187	129	40.52	<.0001
Time1	-163.64	23.2415	503	-7.04	<.0001
Slope26	130.75	26.6265	503	4.91	<.0001

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept at Session=1 Time=0	1961.89	48.4187	129	40.52	<.0001
Intercept at Session=2 Time=1	1798.25	47.0035	115	38.26	<.0001
Intercept at Session=3 Time=2	1765.36	45.9134	104	38.45	<.0001
Intercept at Session=4 Time=3	1732.46	45.5443	101	38.04	<.0001
Intercept at Session=5 Time=4	1699.57	45.9134	104	37.02	<.0001
Intercept at Session=6 Time=5	1666.68	47.0035	115	35.46	<.0001
Rate of change from session 2 to 6	-32.8932	5.8104	503	-5.66	<.0001



### Model 5b: Random Slope, Fixed Deviation Slope Model

Level 1:  $y_{ti} = \beta_{0i} + \beta_{1i}(\text{Time}_{ti}) + \beta_{2i}(\text{Slope26}_{ti}) + e_{ti}$   
 Level 2: Intercept:  $\beta_{0i} = \gamma_{00} + U_{0i}$   
 Time:  $\beta_{1i} = \gamma_{10} + U_{1i}$   
 Slope26:  $\beta_{2i} = \gamma_{20}$

```
TITLE1 "Model 5b: Random Slope, Fixed Deviation Slope Model";
PROC MIXED DATA=&datafile. COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
CLASS PersonID session;
MODEL rt = Time1 Slope26 / SOLUTION DDFM=Satterthwaite;
RANDOM INTERCEPT Time1 / G GCORR V VCORR TYPE=UN SUBJECT=PersonID;
REPEATED session / R TYPE=VC SUBJECT=PersonID;
ODS OUTPUT InfoCrit=FitRand16Fix26; RUN; TITLE1;
```

Iteration History

Iteration	Evaluations	-2 Res Log Like	Criterion
0	1	9188.48345679	
1	1	<b>8333.38121978</b>	0.00000000

Estimated G Matrix

Row	Effect	PersonID	Col1	Col2
1	Intercept	101	254290	-12982
2	Time1	101	-12982	2346.46

Estimated G Correlation Matrix

Row	Effect	PersonID	Col1	Col2
1	Intercept	101	1.0000	-0.5315
2	Time1	101	-0.5315	1.0000

Estimated V Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	<b>280225</b>	241308	228325	215343	202361	189378
2	241308	<b>256606</b>	220036	209400	198764	188128
3	228325	220036	<b>237681</b>	203457	195168	186878
4	215343	209400	203457	<b>223449</b>	191571	185628
5	202361	198764	195168	191571	<b>213909</b>	184378
6	189378	188128	186878	185628	184378	<b>209063</b>

This random slope model predicts the same kind of **V** matrix as would a random linear time model —on the variance side, that’s what this is!

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr >  Z
UN(1,1)	PersonID	254290	37895	6.71	<.0001
UN(2,1)	PersonID	-12982	3620.52	-3.59	0.0003
UN(2,2)	PersonID	2346.46	551.40	4.26	<.0001
Session	PersonID	25934	1827.00	14.20	<.0001

Null Model Likelihood Ratio Test

DF	Chi-Square	Pr > ChiSq
3	855.10	<.0001

Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8333.4	<b>4</b>	8341.4	8341.4	8345.6	8351.8	8355.8

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	1961.89	52.6735	109	37.25	<.0001
Time1	-163.64	20.8345	467	-7.85	<.0001
Slope26	130.75	23.2213	403	5.63	<.0001

Is the random time slope significant?  
How do we know?

```
* Call macro to calculate LRT for nested models;
%FitTest(FitFewer=FitFix16Fix26, FitMore=FitRand16Fix26);
```

Likelihood Ratio Test for FitFix16Fix26 vs. FitRand16Fix26

Name	Neg2Log		AIC	BIC	DevDiff	DFdiff	Pvalue
	Like	Parms					
FitFix16Fix26	8382.7	2	8386.7	8391.9	.	.	.
FitRand16Fix26	8333.4	4	8341.4	8351.8	49.3059	2	1.965E-11

**Model 5c: Random Slope, Random Deviation Slope Model (Equivalent to 4c)**

Level 1:  $y_{ti} = \beta_{0i} + \beta_{1i}(\text{Time}_{ti}) + \beta_{2i}(\text{Slope26}_{ti}) + e_{ti}$

Level 2: Intercept:  $\beta_{0i} = \gamma_{00} + U_{0i}$

Time:  $\beta_{1i} = \gamma_{10} + U_{1i}$

Slope26:  $\beta_{2i} = \gamma_{20} + U_{2i}$

```
TITLE1 "Model 5c: Random Slope, Random Deviation Slope Model";
PROC MIXED DATA=&datafile. COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
CLASS PersonID session;
MODEL rt = Time1 Slope26 / SOLUTION DDFM=Satterthwaite;
RANDOM INTERCEPT Time1 Slope26 / G GCORR V VCORR TYPE=UN SUBJECT=PersonID;
REPEATED session / R TYPE=VC SUBJECT=PersonID;
ODS OUTPUT InfoCrit=FitRand16Rand26; RUN; TITLE1;
```

Iteration History

Iteration	Evaluations	-2 Res Log Like	Criterion
0	1	9188.48345679	
1	1	<b>8275.37431715</b>	0.00000000

Estimated G Matrix

Row	Effect	PersonID	Col1	Col2	Col3
1	Intercept	101	284312	-54270	43626
2	Time1	101	-54270	63954	-65626
3	Slope26	101	43626	-65626	69916

Estimated G Correlation Matrix

Row	Effect	PersonID	Col1	Col2	Col3
1	Intercept	101	1.0000	-0.4025	0.3094
2	Time1	101	-0.4025	1.0000	-0.9814
3	Slope26	101	0.3094	-0.9814	1.0000

Estimated V Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	<b>301985</b>	230042	219399	208755	198111	187467
2	230042	<b>257400</b>	227410	215094	202778	190462
3	219399	227410	<b>235385</b>	208013	198314	188615
4	208755	215094	208013	<b>218604</b>	193850	186768
5	198111	202778	198314	193850	<b>207059</b>	184921
6	187467	190462	188615	186768	184921	<b>200747</b>

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr >  Z
UN(1,1)	PersonID	284312	42731	6.65	<.0001
UN(2,1)	PersonID	-54270	18230	-2.98	0.0029
UN(2,2)	PersonID	63954	13244	4.83	<.0001
UN(3,1)	PersonID	43626	19049	2.29	0.0220
UN(3,2)	PersonID	-65626	14154	-4.64	<.0001
UN(3,3)	PersonID	69916	15434	4.53	<.0001
Session	PersonID	17673	1435.84	12.31	<.0001

Null Model Likelihood Ratio Test

DF	Chi-Square	Pr > ChiSq
6	913.11	<.0001

Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8275.4	7	8289.4	8289.6	8296.8	8307.7	8314.7

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	1961.89	54.6805	100	35.88	<.0001
Time1	-163.64	30.2188	100	-5.42	<.0001
Slope26	130.75	32.5530	100	4.02	0.0001

Is the random deviation slope 26 significant? How do we know?

\* Call macro to calculate LRT for nested models;

```
%FitTest(FitFewer=FitRand16Fix26, FitMore=FitRand16Rand26);
```

Likelihood Ratio Test for FitRand16Fix26 vs. FitRand16Rand26

Name	Neg2Log Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitRand16Fix26	8333.4	4	8341.4	8351.8	.	.	.
FitRand16Rand26	8275.4	7	8289.4	8307.7	58.0069	3	1.5665E-12

95% Random Effect Confidence Intervals that describe the predicted range of individual random effects:


$$\text{Random Effect 95\% CI} = \text{fixed effect} \pm (1.96 * \sqrt{\text{Random Variance}})$$

$$\text{Intercept 95\% CI} = \gamma_{00} \pm (1.96 * \sqrt{\tau_{U_0}^2}) \rightarrow 1,961.9 \pm (1.96 * \sqrt{284,312}) = 917 \text{ to } 3,007$$

$$\text{Time 95\% CI} = \gamma_{10} \pm (1.96 * \sqrt{\tau_{U_1}^2}) \rightarrow -163.6 \pm (1.96 * \sqrt{63,954}) = -659 \text{ to } 322$$

$$\text{Slope26 95\% CI} = \gamma_{20} \pm (1.96 * \sqrt{\tau_{U_2}^2}) \rightarrow 130.8 \pm (1.96 * \sqrt{69,916}) = -338 \text{ to } 649$$

So how did we do? Let's compare model predictions in terms of means (top) and variances (bottom)? Note that models 4c and 5c are equivalent, so only 4c is shown.



### Bonus Material: Testing Absolute Fit when using REML

As shown as Model 0, the saturated means, unstructured variance model is the best-fitting model for each side (means and variances). However, when using REML, we cannot do a model comparison against our random piecewise slopes model, because the models cannot differ in their fixed effects for the  $-2LL$  (LRT) to be valid. Instead, we can test the absolute fit for each side of the model separately.

```
TITLE1 "Model 5c: Random Slope, Random Deviation Slope Model (best model, repeated)";
PROC MIXED DATA=&datafile. COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
  CLASS PersonID session;
  MODEL rt = Time1 Slope26 / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT Time1 Slope26 / G GCORR V VCORR TYPE=UN SUBJECT=PersonID;
  REPEATED session / R TYPE=VC SUBJECT=PersonID;
  ODS OUTPUT InfoCrit=FitRand16Rand26; RUN; TITLE1;
```

The means side can be tested using the *same random piecewise slopes* in each version:

```
TITLE1 "Testing Absolute Fit of the Means Model (Using Random Two-Piece Variance Model)";
PROC MIXED DATA=&datafile. COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
  CLASS PersonID session;
  * Add session to the model to fully saturate the means model;
  MODEL rt = Time1 Slope26 session / SOLUTION CL CHISQ DDFM=Satterthwaite HTYPE=1;
  RANDOM INTERCEPT Time1 Slope26 / G GCORR V VCORR TYPE=UN SUBJECT=PersonID;
  REPEATED session / R RCORR TYPE=VC SUBJECT=PersonID; RUN; TITLE1;
```

Solution for Fixed Effects						
Effect	session	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept		1961.89	54.6805	100	35.88	<.0001
time1		-149.60	71.8219	389	-2.08	0.0379
slope26		114.56	89.4520	373	1.28	0.2011
session	1	0	.	.	.	.
session	2	2.8775	67.2576	300	0.04	0.9659
session	3	-27.2205	49.3536	300	-0.55	0.5817
session	4	-24.4190	32.3095	300	-0.76	0.4504
session	5	0	.	.	.	.
session	6	0	.	.	.	.

Type 1 Tests of Fixed Effects						
Effect	Num DF	Den DF	Chi-Square	F Value	Pr > ChiSq	Pr > F
time1	1	100	70.17	70.17	<.0001	<.0001
slope26	1	100	16.13	16.13	<.0001	0.0001
session	3	300	4.74	1.58	0.1922	0.1946

There are now 6 fixed effects for the 6 means, such that this is now equivalent to a saturated means model (even if the coefficients are largely uninterpretable). SAS figures out how many extra contrasts are needed to saturate and only estimates that many.

Htype=1 is a sequential test in order of entry. The multivariate Wald test for session indicates that the 3 extra contrasts did not improve model fit (which is good news).

The variance side can be tested using the *same fixed piecewise slopes* in each version:

```
TITLE1 "Testing Absolute Fit of the Variance Model (Using Fixed Piecewise Means Model)";
PROC MIXED DATA=&datafile. COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
  CLASS PersonID session;
  MODEL rt = Time1 Slope26 / SOLUTION CL CHISQ DDFM=Satterthwaite;
  REPEATED session / R RCORR TYPE=UN SUBJECT=PersonID;
  ODS OUTPUT InfoCrit=FitRandUN; RUN; TITLE1;
  * Call macro to calculate LRT for nested models;
  %FitTest(FitFewer= FitRand16Rand26, FitMore=FitRandUN);
```

What does the nonsignificant  $p$ -value for this model comparison indicate?

(Other irrelevant output omitted)

#### Likelihood Ratio Test for FitRand16Rand26 vs. FitRandUN

Name	Neg2Log Like	Parms	AIC	BIC	DevDiff	Dfdiff	Pvalue
FitRand16Rand26	8275.4	7	8289.4	8307.7	.	.	.
FitRandUN	8259.6	21	8301.6	8356.5	15.7728	14	<b>0.32744</b>



## (Truly Nonlinear) Exponential Models

Previously we used polynomial and piecewise slopes models to describe change in RT by session; now we use an exponential model instead, which will require learning a new procedure—SAS NLMIXED. REML is not available in NLMIXED, so these models will be estimated using ML instead. Additional options related to estimation are specified below. To illustrate NLMIXED, we begin with two familiar models: an empty means, random intercept model (1b), and a random quadratic time model (3b).

### Model 1b. Empty Means, Random Intercept Model via MIXED and NLMIXED

$$\text{Level 1: } y_{ti} = \beta_{0i} + e_{ti}$$

$$\text{Level 2: Intercept: } \beta_{0i} = \gamma_{00}$$

```
TITLE1 "Model 1b: Empty Means, Random Intercept Model";
PROC MIXED DATA=&datafile. NOCLPRINT COVTEST METHOD=ML;
  CLASS PersonID session;
  MODEL rt = / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT / G V VCORR TYPE=UN SUBJECT=PersonID;
  REPEATED session / R TYPE=VC SUBJECT=PersonID; RUN;
```

In the PROC NLMIXED line below, adaptive Gauss-Hermite Quadrature (METHOD=GAUSS) is used to integrate over random effects (necessary for non-normal outcomes, but not really relevant here given that our random effects and residuals here are still assumed to be normally distributed.). Newton-Raphson optimization (TECH=NEWRAP) is a specific way of finding the top of the likelihood mountain. Finally, we also set stricter gradient convergence criteria (GCONV=1e-12) to help ensure that each parameter is really at the top of its dimension of the likelihood mountain.

```
TITLE1 "Model 1b: Empty Means, Random Intercept Model via NLMIXED";
PROC NLMIXED DATA=&datafile. METHOD=GAUSS TECH=NEWRAP GCONV=1e-12;
  * Must define all parameters to be estimated and provide start values;
  * First line is fixed effects, second line is variances and covariances;
  PARMS fint=1770
        VarU0=198820 VarE=44900;
  * Setting up level-2 equations;
  b0i = fint + U0i;
  * Setting up level-1 equation WITHOUT level-1 residual;
  PredY = (b0i);
  * Telling it which DV, defining level-1 residual;
  * RT is normally distributed with a mean of "PredY" and a variance of "VarE";
  MODEL rt ~ normal(PredY, VarE);
  * Random effects are normally distributed with means=0 and estimated variances;
  RANDOM U0i ~ normal([0],[VarU0]) SUBJECT=PersonID;
  * Asking for ICC and its SE;
  ESTIMATE "ICC" VarU0 / (VarU0 + VarE); RUN;
```

### MIXED OUTPUT:

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr >  Z
UN(1,1)	PersonID	198820	29035	6.85	<.0001
Session	PersonID	44900	2825.63	15.89	<.0001

Fit Statistics	
-2 Log Likelihood	8546.3
AIC (smaller is better)	8552.3
AICC (smaller is better)	8552.4
BIC (smaller is better)	8560.2

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	1770.70	45.1952	101	39.18	<.0001

**NLMIXED OUTPUT:**

Fit Statistics				Parameter Estimates								
-2 Log Likelihood												
AIC (smaller is better)			8546.3									
AICC (smaller is better)			8552.3									
BIC (smaller is better)			8552.4									
			8560.2									
Parameter	Estimate	Standard Error	DF	t Value	Pr >  t	Alpha	Lower	Upper	Gradient			
fint	1770.69	45.1952	100	39.18	<.0001	0.05	1681.02	1860.36	-5.4E-6			
VarU0	198820	29035	100	6.85	<.0001	0.05	141216	256424	2.82E-10			
VarE	44900	2825.64	100	15.89	<.0001	0.05	39294	50506	3.324E-9			
Additional Estimates												
Label	Estimate	Standard Error	DF	t Value	Pr >  t	Alpha	Lower	Upper				
ICC	0.8158	0.02404	100	33.94	<.0001	0.05	0.7681	0.8635				

The "gradient" column provides the partial first derivatives of the LL function with respect to each parameter. You want these to be as close to 0 as possible, otherwise they are not trustworthy (i.e., it's not really at the top of its dimension of the LL mountain).

**Model 3b. Random Quadratic Time Model via MIXED and NLMIXED**

```

TITLE1 "Model 3b: Random Quadratic Time Model";
PROC MIXED DATA=&datafile. NOCLPRINT COVTEST METHOD=ML;
CLASS PersonID session;
MODEL rt = time1 time1*time1 / SOLUTION DDFM=Satterthwaite;
RANDOM INTERCEPT time1 time1*time1 / G V VCORR TYPE=UN SUBJECT=PersonID;
REPEATED session / R TYPE=VC SUBJECT=PersonID; RUN;

TITLE1 "Model 3b: Random Quadratic Time Model via NLMIXED";
PROC NLMIXED DATA=&datafile. METHOD=GAUSS TECH=NEWRAP GCONV=1e-12;
* Must define all parameters to be estimated and provide start values;
* First line is fixed effects, second and third lines are variances and covariances;
PARMS fint=1946 flin=-121 fquad=14
VarU0=273306 CovU10=-35626 VarU1=25438
CovU20=3845 CovU21=-3838 VarU2=622 VarE=20298;
* Setting up level-2 equations;
b0i = fint + U0i;
b1i = flin + U1i;
b2i = fquad + U2i;
* Setting up level-1 equation WITHOUT level-1 residual;
PredY = (b0i) + (b1i*time1) + (b2i*time1*time1);
* Telling it which DV, defining level-1 residual;
* RTs is normally distributed with a mean of "PredY" and a variance of "VarE";
MODEL rt ~ normal(PredY, VarE);
* Random effects are normally distributed with means=0 and estimated variances/covariances;
RANDOM U0i U1i U2i ~ normal([0,0,0],[VarU0,CovU10,VarU1,CovU20,CovU21,VarU2])
SUBJECT=PersonID; RUN;
    
```

**MIXED OUTPUT:**

		Estimated G Matrix				
Row	Effect	PersonID	Col1	Col2	Col3	
1	Intercept	101	273306	-35262	3845.38	
2	time1	101	-35262	25438	-3837.76	
3	time1*time1	101	3845.38	-3837.76	622.81	
Covariance Parameter Estimates						
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr >  Z	
UN(1,1)	PersonID	273306	40828	6.69	<.0001	
UN(2,1)	PersonID	-35262	11765	-3.00	0.0027	
UN(2,2)	PersonID	25438	5781.19	4.40	<.0001	
UN(3,1)	PersonID	3845.38	1920.35	2.00	0.0452	
UN(3,2)	PersonID	-3837.76	968.79	-3.96	<.0001	
UN(3,3)	PersonID	622.81	169.99	3.66	0.0001	
Session	PersonID	20298	1649.11	12.31	<.0001	
Fit Statistics						
-2 Log Likelihood			8321.8			
AIC (smaller is better)			8341.8			
AICC (smaller is better)			8342.1			
BIC (smaller is better)			8367.9			

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	1945.85	53.5825	101	36.32	<.0001
time1	-120.90	19.9481	101	-6.06	<.0001
time1*time1	13.8656	3.3985	101	4.08	<.0001

**NLMIXED OUTPUT:**

Iteration History					
Iter	Calls	NegLogLike	Diff	MaxGrad	Slope
1*	24	4160.89414	0.029504	0.003755	-0.05484
2*	36	4160.89137	0.002776	0.000271	-0.00332
3*	48	4160.88737	0.003992	0.000167	-0.00504
4*	60	4160.88433	0.003045	0.000053	-0.00511
5*	72	<b>4160.8839</b>	0.000433	4.276E-6	-0.00059

NOTE: ABSGCONV convergence criterion satisfied.

Fit Statistics	
-2 Log Likelihood	<b>8321.8</b>
AIC (smaller is better)	8341.8
AICC (smaller is better)	8342.1
BIC (smaller is better)	8367.9

Parameter Estimates									
Parameter	Estimate	Standard Error	DF	t Value	Pr >  t	Alpha	Lower	Upper	Gradient
fint	1945.85	53.5832	98	36.31	<.0001	0.05	1839.52	2052.18	1.81E-10
flin	-120.90	19.9571	98	-6.06	<.0001	0.05	-160.50	-81.2957	8.76E-11
fquad	13.8656	3.3987	98	4.08	<.0001	0.05	7.1210	20.6102	2.24E-10
VarU0	273313	40831	98	6.69	<.0001	0.05	192286	354341	-6.15E-7
CovU10	-35568	11799	98	-3.01	0.0033	0.05	-58982	-12154	-4.28E-6
VarU1	25474	5791.26	98	4.40	<.0001	0.05	13981	36966	-3.04E-6
CovU20	3898.39	1923.78	98	2.03	0.0454	0.05	80.7058	7716.07	1.039E-8
CovU21	-3841.28	969.79	98	-3.96	0.0001	0.05	-5765.79	-1916.77	-3E-7
VarU2	622.95	170.04	98	3.66	0.0004	0.05	285.52	960.38	-2.74E-6
VarE	20299	1649.21	98	12.31	<.0001	0.05	17026	23572	-8.05E-9

Because variances can be hard to estimate, the negative exponential models that follow instead estimate standard deviations (SD), and then calculate variances by squaring those SDs.

**Model 6a. Negative Exponential Model (Fixed Asymptote, Fixed Amount, Fixed Rate)**

```
TITLE1 "Negative Exponential Model via NLMIXED";
TITLE2 "Model 6a: Fixed Asymptote, Fixed Amount, Fixed Rate";
PROC NLMIXED DATA=&datafile. METHOD=GAUSS TECH=NEWWRAP GCONV=1e-12;
* Must define all parameters to be estimated and provide start values;
* First line is fixed effects, second line is variances (in SD metric);
PARMS fasymp= 1600 famount=300 frate=-1
      SDE=600;
* Setting up level-2 equations;
b0i = fasymp;
bli = famount;
b2i = frate;
* Setting up level-1 equation WITHOUT level-1 residual;
PredY = (b0i) + (bli*EXP(b2i*time1));
* Telling it which DV, defining level-1 residual;
* RTs is normally distributed with a mean of "PredY" and a variance of "VarE";
MODEL rt ~ normal(PredY, sde*sde);
* Labeling estimated parameters;
ESTIMATE "Fixed Asymptote"          fasymp;
ESTIMATE "Fixed Amount"            famount;
ESTIMATE "Fixed Rate"              frate;
ESTIMATE "Residual E Variance"     sde*sde;
* Creating extra parameters and predicted means;
ESTIMATE "Fixed Intercept"         fasymp+famount;
```

```
ESTIMATE "Session=1 Time=0 Predicted Mean" fasymp+(famout*EXP(frate*0));
ESTIMATE "Session=2 Time=1 Predicted Mean" fasymp+(famout*EXP(frate*1));
ESTIMATE "Session=3 Time=2 Predicted Mean" fasymp+(famout*EXP(frate*2));
ESTIMATE "Session=4 Time=3 Predicted Mean" fasymp+(famout*EXP(frate*3));
ESTIMATE "Session=5 Time=4 Predicted Mean" fasymp+(famout*EXP(frate*4));
ESTIMATE "Session=6 Time=5 Predicted Mean" fasymp+(famout*EXP(frate*5));
RUN;
```

Iteration History

Iter	Calls	NegLogLike	Diff	MaxGrad	Slope
1*	15	4631.13397	7.008659	0.635591	-1452266
2*	21	4631.1316	0.002371	0.398959	-0.00545
3*	27	4631.1279	0.003703	0.349025	-0.00405
4*	35	4625.36573	5.762165	11.20453	-0.06333
5*	41	4624.07696	1.28877	1.987387	-1.51198
6*	48	4609.76677	14.3102	13.15472	-3.73214
7*	55	4609.06439	0.702377	6.387425	-7.52845
8*	61	4608.1577	0.906689	3.89459	-2.63132
9*	67	4607.16533	0.992366	0.236206	-1.46147
10*	73	4606.91758	0.247757	0.260529	-0.34753
11*	79	4606.77195	0.145622	0.951019	-0.1909
12	86	4606.69057	0.081389	1.227711	-0.61523
13	92	4606.60972	0.080842	1.38279	-0.21767
14	98	4606.57997	0.029751	0.310892	-0.05081
15	104	4606.57696	0.003013	0.044025	-0.00564
16	110	4606.57692	0.000043	0.000582	-0.00008
17	116	4606.57692	1.039E-8	1.704E-7	-2.08E-8

NOTE: ABSGCONV convergence criterion satisfied.

Fit Statistics

-2 Log Likelihood	9213.2
AIC (smaller is better)	9221.2
AICC (smaller is better)	9221.2
BIC (smaller is better)	9238.8

Parameter Estimates

Parameter	Estimate	Standard Error	DF	t Value	Pr >  t	Alpha	Lower	Upper	Gradient
fasymp	1675.25	54.8326	606	30.55	<.0001	0.05	1567.56	1782.93	1.069E-9
famout	284.71	64.5965	606	4.41	<.0001	0.05	157.85	411.57	3.15E-10
frate	-0.6698	0.4247	606	-1.58	0.1153	0.05	-1.5039	0.1643	1.704E-7
SDE	484.28	13.9107	606	34.81	<.0001	0.05	456.97	511.60	-336E-13

Additional Estimates

Label	Estimate	Standard Error	DF	t Value	Pr >  t	Alpha	Lower	Upper
Fixed Asymptote	1675.25	54.8326	606	30.55	<.0001	0.05	1567.56	1782.93
Fixed Amount	284.71	64.5965	606	4.41	<.0001	0.05	157.85	411.57
Fixed Rate	-0.6698	0.4247	606	-1.58	0.1153	0.05	-1.5039	0.1643
Residual E Variance	234532	13474	606	17.41	<.0001	0.05	208071	260992
Fixed Intercept	1959.96	47.8094	606	41.00	<.0001	0.05	1866.07	2053.85
Session=1 Time=0 Predicted Mean	1959.96	47.8094	606	41.00	<.0001	0.05	1866.07	2053.85
Session=2 Time=1 Predicted Mean	1820.97	36.2937	606	50.17	<.0001	0.05	1749.69	1892.24
Session=3 Time=2 Predicted Mean	1749.83	30.7816	606	56.85	<.0001	0.05	1689.38	1810.28
Session=4 Time=3 Predicted Mean	1713.42	23.4669	606	73.01	<.0001	0.05	1667.33	1759.51
Session=5 Time=4 Predicted Mean	1694.79	27.9049	606	60.73	<.0001	0.05	1639.98	1749.59
Session=6 Time=5 Predicted Mean	1685.25	36.3828	606	46.32	<.0001	0.05	1613.80	1756.70

### Model 6b. Negative Exponential Model (Add Random Asymptote, Fixed Amount, Fixed Rate)

```
TITLE1 "Negative Exponential Model via NLMIXED";
TITLE2 "Model 6b: Add Random Asymptote, Fixed Amount, Fixed Rate";
PROC NLMIXED DATA=&datafile. METHOD=GAUSS TECH=NEWWRAP GCONV=1e-12;
* Must define all parameters to be estimated and provide start values;
* First line is fixed effects, second line is variances (in SD metric);
PARMS fasymp= 1675 famout=284 frate=-.7
      sdE=474 sdU0=10;
```

```

* Setting up level-2 equations;
  b0i = fasymp + U0i;
  bli = famount;
  b2i = frate;
* Setting up level-1 equation WITHOUT level-1 residual;
  PredY = (b0i) + (bli*EXP(b2i*time1));
* Telling it which DV, defining level-1 residual;
* RTs is normally distributed with a mean of "PredY" and a variance of "VarE";
  MODEL rt ~ normal(PredY, sdE*sdE);
* Defining random effects: normally distributed with means and variances;
  RANDOM U0i ~ normal([0],[sdU0*sdU0]) SUBJECT=PersonID;
* Labeling estimated parameters;
  ESTIMATE "Fixed Asymptote"           fasymp;
  ESTIMATE "Fixed Intercept"          fasymp+famount;
  ESTIMATE "Fixed Amount"            famount;
  ESTIMATE "Fixed Rate"              frate;
  ESTIMATE "Residual E Variance"     sdE*sdE;
  ESTIMATE "Random Asymptote U0 Variance" sdU0*sdU0; RUN;

```

Iteration History

Iter	Calls	NegLogLike	Diff	MaxGrad	Slope
1*	17	4284.53992	321.7295	2.366279	-111.108
2*	25	4260.22207	24.31785	0.670259	-67.5009
3*	37	4206.38115	53.84092	3.324296	-8.69485
4	44	4203.31763	3.063522	0.287907	-10.2693
5	51	4202.06928	1.248343	0.032384	-2.22851
6	58	4202.01577	0.053511	0.001224	-0.10347
7	65	<b>4202.01562</b>	0.00015	2.948E-6	-0.0003

NOTE: ABSGCONV convergence criterion satisfied.

Fit Statistics

Statistic	Value
-2 Log Likelihood	8404.0
AIC (smaller is better)	8414.0
AICC (smaller is better)	8414.1
BIC (smaller is better)	8427.1

Is the random asymptote variance significant?

Parameter Estimates

Parameter	Estimate	Standard Error	DF	t Value	Pr >  t	Alpha	Lower	Upper	Gradient
fasymp	1675.25	49.2032	100	34.05	<.0001	0.05	1577.63	1772.87	3.31E-8
famount	284.71	24.5497	100	11.60	<.0001	0.05	236.00	333.41	1.102E-8
frate	-0.6698	0.1614	100	-4.15	<.0001	0.05	-0.9900	-0.3495	2.948E-6
sdE	184.05	5.7913	100	31.78	<.0001	0.05	172.56	195.54	-1.55E-7
sdU0	447.95	32.4064	100	13.82	<.0001	0.05	383.65	512.24	-1.59E-6

Additional Estimates

Label	Estimate	Standard Error	DF	t Value	Pr >  t	Alpha	Lower	Upper
Fixed Asymptote	1675.25	49.2032	100	34.05	<.0001	0.05	1577.63	1772.87
Fixed Intercept	1959.96	48.1335	100	40.72	<.0001	0.05	1864.46	2055.45
Fixed Amount	284.71	24.5497	100	11.60	<.0001	0.05	236.00	333.41
Fixed Rate	-0.6698	0.1614	100	-4.15	<.0001	0.05	-0.9900	-0.3495
Residual E Variance	33875	2131.79	100	15.89	<.0001	0.05	29645	38104
Random Asymptote U0 Variance	200656	29033	100	6.91	<.0001	0.05	143056	258256

### Model 6c. Negative Exponential Model (Random Asymptote, Add Random Amount, Fixed Rate)

```

TITLE1 "Negative Exponential Model via NLMIXED";
TITLE2 "Model 6c: Random Asymptote, Add Random Amount, Fixed Rate";
PROC NLMIXED DATA=&datafile. METHOD=GAUSS TECH=NEWRAP GCONV=1e-12;
* Must define all parameters to be estimated and provide start values;
* First line is fixed effects, second line is variances and covariances (in SD metric);
  PARMS fasymp=1675 famount=284 frate=-.7
        sdE=184 sdU0=447 sdU01=1 sdU1=10;
* Setting up level-2 equations;
  b0i = fasymp + U0i;
  bli = famount + U1i;

```

```

b2i = frate;
* Setting up level-1 equation WITHOUT level-1 residual;
PredY = (b0i) + (bli*EXP(b2i*time1));
* Telling it which DV, defining level-1 residual;
* RTs is normally distributed with a mean of "y" and a variance of "VarE";
MODEL rt ~ normal(PredY, sdE*sdE);
* Defining random effects: normally distributed with means and variances;
RANDOM U0i U1i ~ normal([0,0],[sdU0*sdU0,sdU01*sdU01,sdU1*sdU1]) SUBJECT=PersonID;
* Labeling estimated parameters;
ESTIMATE "Fixed Asymptote"          fasymp;
ESTIMATE "Fixed Amount"             famount;
ESTIMATE "Fixed Rate"               frate;
ESTIMATE "Residual E Variance"      sdE*sdE;
ESTIMATE "Random Asymptote U0 Variance" sdU0*sdU0;
ESTIMATE "Asymptote-Amount U01 Covariance" sdU01*sdU01;
ESTIMATE "Random Amount U1 Variance" sdU1*sdU1;
ESTIMATE "Asymptote-Amount Correlation" (sdU01*sdU01)/(sdU0*sdU1); RUN;
    
```

NOTE: ABSGCONV convergence criterion satisfied.

Fit Statistics

-2 Log Likelihood	8327.3
AIC (smaller is better)	8341.3
AICC (smaller is better)	8341.5
BIC (smaller is better)	8359.6

Is the random amount variance significant?

Parameter Estimates

Parameter	Estimate	Standard Error	DF	t Value	Pr >  t	Alpha	Lower	Upper	Gradient
fasymp	1683.48	45.4523	99	37.04	<.0001	0.05	1593.30	1773.67	-3.75E-6
famount	279.94	33.5457	99	8.35	<.0001	0.05	213.38	346.51	-1.59E-7
frate	-0.7533	0.1181	99	-6.38	<.0001	0.05	-0.9877	-0.5189	5.535E-6
sdE	151.79	5.3422	99	28.41	<.0001	0.05	141.19	162.39	-2.98E-7
sdU0	436.83	31.8997	99	13.69	<.0001	0.05	373.54	500.13	-6.18E-7
sdU01	81.5157	90.6707	99	0.90	0.3708	0.05	-98.3947	261.43	-3.39E-6
sdU1	277.95	28.4727	99	9.76	<.0001	0.05	221.45	334.44	-1.86E-7

Additional Estimates

Label	Estimate	Standard Error	DF	t Value	Pr >  t	Alpha	Lower	Upper
Fixed Asymptote	1683.48	45.4523	99	37.04	<.0001	0.05	1593.30	1773.67
Fixed Amount	279.94	33.5457	99	8.35	<.0001	0.05	213.38	346.51
Fixed Rate	-0.7533	0.1181	99	-6.38	<.0001	0.05	-0.9877	-0.5189
Residual E Variance	23039	1621.75	99	14.21	<.0001	0.05	19821	26257
Random Asymptote U0 Variance	190823	27870	99	6.85	<.0001	0.05	135523	246122
Asymptote-Amount U01 Covariance	6644.80	14782	99	0.45	0.6540	0.05	-22686	35976
Random Amount U1 Variance	77254	15828	99	4.88	<.0001	0.05	45848	108659
Asymptote-Amount Correlation	0.05473	0.1224	99	0.45	0.6557	0.05	-0.1881	0.2975

95% Random Effect Confidence Intervals that describe the predicted range of individual random effects:

Random Effect 95% CI = fixed effect ± (1.96\*√Random Variance)

Asymptote 95% CI =  $\gamma_{00} \pm (1.96 * \sqrt{\tau_{U_0}^2}) \rightarrow 1,683.5 \pm (1.96 * \sqrt{190,823}) = 827 \text{ to } 2,540$

Amount 95% CI =  $\gamma_{10} \pm (1.96 * \sqrt{\tau_{U_1}^2}) \rightarrow 279.9 \pm (1.96 * \sqrt{77,254}) = -265 \text{ to } 825$


**Model 6d. Negative Exponential Model (Random Asymptote and Amount, Add Random Rate)**

**From the log:**  
 ERROR: Quadrature accuracy of 0.000100 could not be achieved with 31 points. The achieved accuracy was 1.000000.

No convergence, after several tries with different start values and relaxing the estimation options...  
 \* **Simpler estimation to get start values;**  
**PROC NL MIXED DATA=&datafile. METHOD=FIRO;**

Since the random rate parameter will not estimate, we will call this “as done as we can be”.

So how did we do? Let's compare model predictions in terms of means (top) and variances (bottom)?



As with the polynomial and piecewise models, we can test the absolute fit of each side of the model. **To test the means side, we need to manually create the three extra session-specific contrasts and then request a multivariate Wald Test for their significance, as follows:**

```
TITLE1 'Test Absolute Fit of Means Model using Random Asymptote and Amount Variance Model';
PROC NL MIXED DATA=&datafile. METHOD=GAUSS TECH=NEWWRAP GCONV=1e-12;
* Programming statements to create session contrasts;
  IF Session=4 THEN s4=1; ELSE s4=0;
  IF Session=5 THEN s5=1; ELSE s5=0;
  IF Session=6 THEN s6=1; ELSE s6=0;
* Must define all parameters to be estimated and provide start values;
* First line is fixed effects, second line is variances (in SD metric);
  PARS fasymp=1675 famount=284 frate=-.7 fs4=0 fs5=0 fs6=0
       sdE=184 sdU0=447 sdU01=1 sdU1=10;
* Setting up level-2 equations;
  b0i = fasymp + U0i;
```

```

b1i = famount + U1i;
b2i = frate;
* Setting up level-1 equation WITHOUT level-1 residual;
* Adding session-specific contrasts;
PredY = (b0i) + (b1i*EXP(b2i*time1)) + (fs4*s4) + (fs5*s5) + (fs6*s6);
* Telling it which DV, defining level-1 residual;
* RTs is normally distributed with a mean of 'y' and a variance of 'VarE';
MODEL rt ~ normal(PredY, sdE*sdE);
* Defining random effects: normally distributed with means and variances;
RANDOM U0i U1i ~ normal([0,0],[sdU0*sdU0,sdU01*sdU01,sdU1*sdU1]) SUBJECT=PersonID;
* Multivariate Wald test for extra session contrasts;
CONTRAST "Test of Absolute Fit of Means Model" fs4*1, fs5*1, fs6*1; RUN;

```

Parameter Estimates					
Parameter	Estimate	Standard Error	DF	t Value	Pr >  t
fasymp	1711.96	52.4575	99	32.64	<.0001
famount	250.76	43.4096	99	5.78	<.0001
frate	-0.9161	0.2096	99	-4.37	<.0001
fs4	-10.2218	24.4668	99	-0.42	0.6770
fs5	-11.2088	28.2634	99	-0.40	0.6925
fs6	-42.3943	30.5202	99	-1.39	0.1679

There are now 6 fixed effects for the 6 means, such that this is now equivalent to a saturated means model (even if the coefficients are largely uninterpretable).

The multivariate Wald test for session indicates that the 3 extra contrasts did not improve model fit (which is good news).

Contrasts				
Label	Num DF	Den DF	F Value	Pr > F
Test of Absolute Fit of Means Model	3	99	0.98	<b>0.4044</b>

Fit Statistics from Exponential Variance Model (4 parms)

-2 Log Likelihood	8324.6
AIC (smaller is better)	8344.6
AICC (smaller is better)	8345.0
BIC (smaller is better)	8370.8

Given that this is now a saturated means model, we can compare the fit of its exponential variance model (random asymptote + random amount) to an Unstructured R model estimated in ML instead of REML in MIXED.

```

TITLE1 'Ch 6: 0: Saturated Means, Unstructured Variance Model';
TITLE2 'Using ML Instead of REML';
PROC MIXED DATA=&datafile. COVTEST NOCLPRINT NAMELEN=100 IC METHOD=ML;
CLASS PersonID session;
MODEL rt = session / SOLUTION CL CHISQ DDFM=Satterthwaite;
REPEATED session / TYPE=UN SUBJECT=PersonID;
ODS OUTPUT InfoCrit=FitMLSatUN; RUN; TITLE1; TITLE2;

```

Fit Statistics from UN Variance Model (21 parms)

-2 Log Likelihood	8278.1
AIC (Smaller is Better)	8332.1
AICC (Smaller is Better)	8334.7
BIC (Smaller is Better)	8402.7

$-2\Delta LL(17) = 46.5, p < .001$ , so the variance model does not fit.

### Bonus Material: Using Log Time to Mimic Exponential Models

Previously we used SAS NLMIXED to estimate a truly nonlinear model with an exponential curve, in which REML is not available. Below we examine a strategy to mimic the same shape that uses linear and quadratic effects of natural-log-transformed time instead, such that we can use REML in MIXED.

#### SAS Syntax for Data Manipulation:

```

* Log-transforming time;
DATA &datafile.; SET &datafile.;
logtime = LOG(session); LABEL logtime = "logtime: Log-Transformed Time (0=1)"; RUN;

```



### Model 7a. Fixed Linear Log Time, Random Intercept Model

Level 1:  $y_{ti} = \beta_{0i} + \beta_{1i}(\text{LogTime}_{ti}) + e_{ti}$   
 Level 2: Intercept:  $\beta_{0i} = \gamma_{00} + U_{0i}$   
 Linear LogTime:  $\beta_{1i} = \gamma_{10}$

```
TITLE1 'Bonus: 7a: Fixed Linear Log Time, Random Intercept Model';
PROC MIXED DATA=&datafile. COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
  CLASS PersonID session;
  MODEL rt = logtime / SOLUTION CHISQ DDFM=Satterthwaite;
  RANDOM INTERCEPT / TYPE=UN SUBJECT=PersonID;
  REPEATED session / TYPE=VC SUBJECT=PersonID;
  ODS OUTPUT CovParms=CovFixLinLog InfoCrit=FitFixLinLog;
  ESTIMATE 'Intercept at Session=1' intercept 1 logtime 0;
  ESTIMATE 'Intercept at Session=2' intercept 1 logtime 0.6931;
  ESTIMATE 'Intercept at Session=3' intercept 1 logtime 1.0986;
  ESTIMATE 'Intercept at Session=4' intercept 1 logtime 1.3863;
  ESTIMATE 'Intercept at Session=5' intercept 1 logtime 1.6094;
  ESTIMATE 'Intercept at Session=6' intercept 1 logtime 1.7918;
RUN; TITLE1;
```

Estimated V Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	<b>236852</b>	202669	202669	202669	202669	202669
2	202669	<b>236852</b>	202669	202669	202669	202669
3	202669	202669	<b>236852</b>	202669	202669	202669
4	202669	202669	202669	<b>236852</b>	202669	202669
5	202669	202669	202669	202669	<b>236852</b>	202669
6	202669	202669	202669	202669	202669	<b>236852</b>

The predicted V matrix still has a compound symmetry pattern because we have not yet added to the model for the variance (still a random intercept variance only in G).

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z	Pr > Z
UN(1,1)	PersonID	202669	29470	6.88	<.0001
session	PersonID	34183	2153.35	15.87	<.0001

Null Model Likelihood Ratio Test

DF	Chi-Square	Pr > ChiSq
1	805.80	<.0001

Information Criteria


Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8391.3	2	8395.3	8395.3	8397.4	8400.5	8402.5

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	1942.55	47.4172	119	40.97	<.0001
logtime	-156.72	12.4160	504	-12.62	<.0001

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept at Session=1	1942.55	47.4172	119	40.97	<.0001
Intercept at Session=2	1833.93	45.6960	102	40.13	<.0001
Intercept at Session=3	1770.38	45.4206	100	38.98	<.0001
Intercept at Session=4	1725.29	45.5629	101	37.87	<.0001
Intercept at Session=5	1690.33	45.8648	104	36.85	<.0001
Intercept at Session=6	1661.74	46.2336	107	35.94	<.0001



The linear effect of log time mimics an exponential curve (and appears to fit pretty well to the saturated means).

```
* Call macro to calculate pseudo R2;
%PseudoR2(NCov=2, CovFewer=CovEmpty, CovMore=CovFixLinLog);
```

**PseudoR2 (% Reduction) for CovEmpty vs. CovFixLinLog**

Name	CovParm	Subject	Estimate	StdErr	ZValue	ProbZ	PseudoR2
CovEmpty	UN(1,1)	PersonID	200883	29471	6.82	<.0001	.
CovEmpty	session	PersonID	44900	2825.63	15.89	<.0001	.
CovFixLinLog	UN(1,1)	PersonID	202669	29470	6.88	<.0001	-0.00889
CovFixLinLog	session	PersonID	34183	2153.35	15.87	<.0001	<b>0.23868</b>

**Model 7b. Random Linear Log Time**

Level 1:  $y_{ti} = \beta_{0i} + \beta_{1i} (\text{LogTime}_{ti}) + e_{ti}$

Level 2: Intercept:  $\beta_{0i} = \gamma_{00} + U_{0i}$

Linear LogTime:  $\beta_{1i} = \gamma_{10} + U_{1i}$

```
TITLE1 'Bonus: 7b: Random Linear Log Time Model';
PROC MIXED DATA=&datafile. COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
CLASS PersonID session;
MODEL rt = logtime / SOLUTION CHISQ DDFM=Satterthwaite;
RANDOM INTERCEPT logtime / G GCORR V TYPE=UN SUBJECT=PersonID;
REPEATED session / TYPE=VC SUBJECT=PersonID;
ODS OUTPUT InfoCrit=FitRandLinLog; RUN; TITLE1;
```

Estimated G Matrix

Row	Effect	PersonID	Col1	Col2
1	Intercept	101	274252	-44541
2	logtime	101	-44541	23101

Estimated G Correlation Matrix

Row	Effect	PersonID	Col1	Col2
1	Intercept	101	1.0000	-0.5596
2	logtime	101	-0.5596	1.0000

Estimated V Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	<b>298372</b>	243378	225318	212505	202566	194445
2	243378	<b>247724</b>	212036	203829	197463	192262
3	225318	212036	<b>228387</b>	198754	194478	190984
4	212505	203829	198754	<b>219274</b>	192360	190078
5	202566	197463	194478	192360	<b>214838</b>	189375
6	194445	192262	190984	190078	189375	<b>212922</b>

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr >  Z
UN(1,1)	PersonID	274252	41240	6.65	<.0001
UN(2,1)	PersonID	-44541	11493	-3.88	0.0001
UN(2,2)	PersonID	23101	4882.26	4.73	<.0001
session	PersonID	24120	1697.11	14.21	<.0001

Null Model Likelihood Ratio Test

DF	Chi-Square	Pr > ChiSq
3	873.64	<.0001

Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8323.4	4	8331.4	8331.5	8335.7	8341.9	8345.9

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	1942.55	53.7211	100	36.16	<.0001
logtime	-156.72	18.3711	100	-8.53	<.0001

```
* Call macro to calculate LRT for nested models;
%FitTest(FitFewer=FitFixLinLog, FitMore=FitRandLinLog);
```

Likelihood Ratio Test for FitFixLinLog vs. FitRandLinLog

Name	Neg2LogLike	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitFixLinLog	8391.3	2	8395.3	8400.5	.	.	.
FitRandLinLog	8323.4	4	8331.4	8341.9	67.8448	2	<b>1.8874E-15</b>

95% Random Effects Confidence Intervals that describe the predicted range of individual random effects:

Random Effect 95% CI = fixed effect ± (1.96\*√Random Variance)

Intercept 95% CI =  $\gamma_{00} \pm (1.96 * \sqrt{\tau_{U_0}^2}) \rightarrow 1,942.55 \pm (1.96 * \sqrt{274,252}) = 916 \text{ to } 2,969$

Linear LogTime Slope 95% CI =  $\gamma_{10} \pm (1.96 * \sqrt{\tau_{U_1}^2}) \rightarrow -156.72 \pm (1.96 * \sqrt{23,101}) = -455 \text{ to } 141$

As with the previous models, we can test the absolute fit of each side of the model.

**The means side can be tested using the same random linear log time slope in each version:**

```
TITLE1 'Testing Absolute Fit of the Means Model (Using Random Linear Log Time Variance Model)';
PROC MIXED DATA=&datafile. COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
CLASS PersonID session;
MODEL rt = logtime session / SOLUTION CHISQ DDFM=Satterthwaite HTYPE=1;
RANDOM INTERCEPT logtime / V TYPE=UN SUBJECT=PersonID;
REPEATED session / TYPE=VC SUBJECT=PersonID; RUN; TITLE1;
```

Solution for Fixed Effects

Effect	Session	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept		2016.49	210.04	449	9.60	<.0001
logtime		-192.19	120.43	418	-1.60	0.1113
session	1	-54.5944	204.05	400	-0.27	0.7892
session	2	-68.1019	121.83	400	-0.56	0.5765
session	3	-55.3148	74.3533	400	-0.74	0.4573
session	4	-32.2644	42.0226	400	-0.77	0.4431
session	5	0	.	.	.	.
session	6	0	.	.	.	.

SAS figures out how many extra contrasts are needed to saturate the means and only estimates that many (4 here).  
  
The multivariate Wald test for session indicates that the 4 extra contrasts did not improve model fit (which is good news).

Type 1 Tests of Fixed Effects						
Effect	Num DF	Den DF	Chi-Square	F Value	Pr > ChiSq	Pr > F
logtime	1	100	72.77	72.77	<.0001	<.0001
session	4	400	6.69	1.67	0.1531	<b>0.1554</b>


The variance side can be tested using the *same fixed linear log time slope* in each version:

```
TITLE1 'Testing Absolute Fit of the Variance Model (Using Fixed Linear Log Time Means Model)';
PROC MIXED DATA=&datafile. COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
  CLASS PersonID session;
  MODEL rt = logtime / SOLUTION CHISQ DDFM=Satterthwaite;
  REPEATED session / R RCORR TYPE=UN SUBJECT=PersonID;
  ODS OUTPUT CovParms=CovRandLinLog InfoCrit=FitRandUN;
RUN; TITLE1;
* Call macro to calculate LRT for nested models;
  %FitTest(FitFewer=FitRandLinLog, FitMore=FitRandUN);
```

What does this significant *p*-value for this model comparison indicate?

Likelihood Ratio Test for FitRandLinLog vs. FitRandUN

Name	Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitRandLinLog	8323.4	4	8331.4	8341.9	.	.	.
FitRandUN	8265.4	21	8307.4	8362.3	58.0097	17	<b>.000002230</b>



The random linear effect of log time approximates most of the observed variances from the unstructured model, but not well enough (and the same must be true of the covariances, not shown here).

Model 7c. Fixed Quadratic, Random Linear Log Time

Level 1:  $y_{ti} = \beta_{0i} + \beta_{1i} (\text{LogTime}_{ti}) + \beta_{2i} (\text{LogTime}_{ti})^2 + e_{ti}$

Level 2: Intercept:  $\beta_{0i} = \gamma_{00} + U_{0i}$

Linear LogTime:  $\beta_{1i} = \gamma_{10} + U_{1i}$

Quadratic LogTime:  $\beta_{2i} = \gamma_{20}$

```
TITLE1 'Bonus: 7c: Fixed Quadratic, Random Linear Log Time Model';
PROC MIXED DATA=&datafile. COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
  CLASS PersonID session;
  MODEL rt = logtime logtime*logtime / SOLUTION CHISQ DDFM=Satterthwaite;
  RANDOM INTERCEPT logtime / V TYPE=UN SUBJECT=PersonID;
  REPEATED session / TYPE=VC SUBJECT=PersonID;
  ODS OUTPUT CovParms=CovFixQuadLog InfoCrit=FitFixQuadLog;
ESTIMATE 'Intercept at Session=1' intercept 1 logtime 0 logtime*logtime 0;
ESTIMATE 'Intercept at Session=2' intercept 1 logtime 0.6931 logtime*logtime 0.4805;
ESTIMATE 'Intercept at Session=3' intercept 1 logtime 1.0986 logtime*logtime 1.2069;
ESTIMATE 'Intercept at Session=4' intercept 1 logtime 1.3863 logtime*logtime 1.9218;
```

```
ESTIMATE 'Intercept at Session=5' intercept 1 logtime 1.6094 logtime*logtime 2.5903;
ESTIMATE 'Intercept at Session=6' intercept 1 logtime 1.7918 logtime*logtime 3.2104;
RUN; TITLE1;
```

Estimated V Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	<b>298292</b>	243482	225365	212511	202540	194394
2	243482	<b>247510</b>	212083	203861	197483	192273
3	225365	212083	<b>228152</b>	198801	194525	191032
4	212511	203861	198801	<b>219050</b>	192426	190151
5	202540	197483	194525	192426	<b>214637</b>	189468
6	194394	192273	191032	190151	189468	<b>212749</b>

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr >  Z
UN(1,1)	PersonID	274453	41239	6.66	<.0001
UN(2,1)	PersonID	-44682	11493	-3.89	0.0001
UN(2,2)	PersonID	23229	4880.99	4.76	<.0001
session	PersonID	23839	1679.38	14.20	<.0001

Null Model Likelihood Ratio Test

DF	Chi-Square	Pr > ChiSq
3	876.51	<.0001

Information Criteria


Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8309.9	4	8317.9	8318.0	8322.1	8328.4	8332.4

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	1960.96	54.2653	104	36.14	<.0001
logtime	-240.61	39.4593	502	-6.10	<.0001
logtime*logtime	46.9146	19.5288	403	2.40	0.0167

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept at Session=1	1960.96	54.2653	104	36.14	<.0001
Intercept at Session=2	1816.74	48.1936	105	37.70	<.0001
Intercept at Session=3	1753.25	45.9683	105	38.14	<.0001
Intercept at Session=4	1717.57	44.6261	101	38.49	<.0001
Intercept at Session=5	1695.25	44.2821	100	38.28	<.0001
Intercept at Session=6	1680.45	44.9706	106	37.37	<.0001



The quadratic effect of log time appears to make the predicted line a little more bendy, such that it fits the saturated means slightly (but significantly) better.

```
* Call macro to calculate pseudo R2;
  %PseudoR2(NCov=4, CovFewer=CovRandLinLog, CovMore=CovFixLinLog);
```

**PseudoR2 (% Reduction) for CovRandLinLog vs. CovFixQuadLog**

Name	CovParm	Subject	Estimate	StdErr	ZValue	ProbZ	PseudoR2
CovRandLinLog	UN(1,1)	PersonID	274252	41240	6.65	<.0001	.
CovRandLinLog	UN(2,2)	PersonID	23101	4882.26	4.73	<.0001	.
CovRandLinLog	session	PersonID	24120	1697.11	14.21	<.0001	.
CovFixQuadLog	UN(1,1)	PersonID	274453	41239	6.66	<.0001	-0.000733
CovFixQuadLog	UN(2,2)	PersonID	23229	4880.99	4.76	<.0001	-0.005551
CovFixQuadLog	session	PersonID	23839	1679.38	14.20	<.0001	<b>0.011672</b>

**Model 7d. Random Quadratic Log Time**

Level 1:  $y_{ti} = \beta_{0i} + \beta_{1i}(\text{LogTime}_{ti}) + \beta_{2i}(\text{LogTime}_{ti})^2 + e_{ti}$

Level 2: Intercept:  $\beta_{0i} = \gamma_{00} + U_{0i}$

Linear LogTime:  $\beta_{1i} = \gamma_{10} + U_{1i}$

Quadratic LogTime:  $\beta_{2i} = \gamma_{20} + U_{2i}$

```
TITLE1 'Bonus: 7d: Random Quadratic Log Time Model';
PROC MIXED DATA=&datafile. COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
  CLASS PersonID session;
  MODEL rt = logtime logtime*logtime / SOLUTION CHISQ DDFM=Satterthwaite;
  RANDOM INTERCEPT logtime logtime*logtime / V TYPE=UN SUBJECT=PersonID;
  REPEATED session / TYPE=VC SUBJECT=PersonID;
  ODS OUTPUT InfoCrit=FitRandQuadLog; RUN; TITLE1;
```

Estimated G Matrix

Row	Effect	PersonID	Col1	Col2	Col3
1	Intercept	101	285072	-80403	15959
2	logtime	101	-80403	199200	-86830
3	logtime*logtime	101	15959	-86830	43025

Estimated G Correlation Matrix

Row	Effect	PersonID	Col1	Col2	Col3
1	Intercept	101	1.0000	-0.3374	0.1441
2	logtime	101	-0.3374	1.0000	-0.9379
3	logtime*logtime	101	0.1441	-0.9379	1.0000

Estimated V Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	<b>302304</b>	237009	216002	204280	197007	192244
2	237009	<b>253981</b>	226105	213856	201669	189971
3	216002	226105	<b>236995</b>	209779	198899	187976
4	204280	213856	209779	<b>219785</b>	194472	186263
5	197007	201669	198899	194472	<b>206863</b>	184764
6	192244	189971	187976	186263	184764	<b>200661</b>

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z	Pr >  Z
UN(1,1)	PersonID	285072	42684	6.68	<.0001
UN(2,1)	PersonID	-80403	31879	-2.52	0.0117
UN(2,2)	PersonID	199200	42606	4.68	<.0001
UN(3,1)	PersonID	15959	14893	1.07	0.2839
UN(3,2)	PersonID	-86830	20318	-4.27	<.0001
UN(3,3)	PersonID	43025	10274	4.19	<.0001
session	PersonID	17232	1399.97	12.31	<.0001

```
Null Model Likelihood Ratio Test
DF      Chi-Square      Pr > ChiSq
6        920.29          <.0001
```

```
Information Criteria
Neg2LogLike  Parms      AIC      AICC      HQIC      BIC      CAIC
8266.1       7          8280.1   8280.3   8287.5   8298.4   8305.4
```

```
Solution for Fixed Effects
Standard
Effect      Estimate      Error      DF      t Value      Pr > |t|
Intercept   1960.96      54.6521   100     35.88        <.0001
logtime     -240.61      54.1434   100     -4.44        <.0001
logtime*logtime  46.9146     26.4888   100     1.77         0.0796
```

```
* Call macro to calculate LRT for nested models;
%FitTest(FitFewer=FitFixQuadLog, FitMore=FitRandQuadLog);
```

#### Likelihood Ratio Test for FitFixQuadLog vs. FitRandQuadLog

Name	Neg2Log Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitFixQuadLog	8309.9	4	8317.9	8328.4	.	.	.
FitRandQuadLog	8266.1	7	8280.1	8298.4	43.7825	3	<b>1.6786E-9</b>

#### 95% Random Effect Confidence Intervals that describe the predicted range of individual random effects:

Random Effect 95% CI = fixed effect  $\pm (1.96 * \sqrt{\text{Random Variance}})$

Intercept 95% CI =  $\gamma_{00} \pm (1.96 * \sqrt{\tau_{U_0}^2}) \rightarrow 1,960.96 \pm (1.96 * \sqrt{285,072}) = 914 \text{ to } 3,007$

Linear LogTime Slope 95% CI =  $\gamma_{10} \pm (1.96 * \sqrt{\tau_{U_1}^2}) \rightarrow -240.61 \pm (1.96 * \sqrt{199,200}) = -115 \text{ to } 634$

Quadratic LogTime Slope 95% CI =  $\gamma_{20} \pm (1.96 * \sqrt{\tau_{U_2}^2}) \rightarrow 46.91 \pm (1.96 * \sqrt{43025}) = -360 \text{ to } 453$

As with the previous models, we can test the absolute fit of each side of the model.

**The means side can be tested using the same random quadratic log time slope in each version. Using session to do so resulted in non-convergence, so I manually created 3 contrasts instead.**

```
* Create session contrasts;
DATA &datafile.; SET &datafile.;
  IF Session=4 THEN s4=1; ELSE s4=0;
  IF Session=5 THEN s5=1; ELSE s5=0;
  IF Session=6 THEN s6=1; ELSE s6=0; RUN;

TITLE1 'Testing Absolute Fit of the Means Model (Using Random Quadratic Log Time Variance Model)';
PROC MIXED DATA=&datafile. COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
  CLASS PersonID session;
  MODEL rt = logtime logtime*logtime s4 s5 s6 / SOLUTION CHISQ DDFM=Satterthwaite;
  RANDOM INTERCEPT logtime logtime*logtime / V TYPE=UN SUBJECT=PersonID;
  REPEATED session / TYPE=VC SUBJECT=PersonID;
* Multivariate Wald test for extra session contrasts;
  CONTRAST "Test of Absolute Fit of Means Model" s4 1, s5 1, s6 1; RUN; TITLE1;
```

Type 3 Tests of Fixed Effects						
Effect	Num DF	Den DF	Chi-Square	F Value	Pr > ChiSq	Pr > F
logtime	1	305	9.99	9.99	0.0016	0.0017
logtime*logtime	1	392	0.57	0.57	0.4486	0.4491
s4	1	300	0.02	0.02	0.8970	0.8971
s5	1	300	0.09	0.09	0.7705	0.7707
s6	1	300	0.00	0.00	0.9818	0.9818

The multivariate Wald test for session indicates that the 3 extra contrasts did not improve model fit (which is good news).

Contrasts						
Label	Num DF	Den DF	Chi-Square	F Value	Pr > ChiSq	Pr > F
Test of Absolute Fit of Means Model	3	300	1.31	0.44	0.7257	0.7259

**The variance side can be tested using the *same fixed quadratic log time slope* in each version:**

```
TITLE1 'Bonus: Testing Absolute Fit of the Variance Model (Using Fixed Quadratic Log Time Means Model)'; PROC MIXED DATA=&datafile. COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
CLASS PersonID session;
MODEL rt = logtime*logtime / SOLUTION CHISQ DDFM=Satterthwaite;
REPEATED session / R RCORR TYPE=UN SUBJECT=PersonID;
ODS OUTPUT InfoCrit=FitRandUN; RUN; TITLE1;

* Call macro to calculate LRT for nested models;
%FitTest(FitFewer=FitRandQuadLog, FitMore=FitRandUN);
```

What does this significant *p*-value for this model comparison indicate?

**Likelihood Ratio Test for FitRandQuadLog vs. FitRandUN**

Name	Neg2Log						
	Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitRandQuadLog	8266.1	7	8280.1	8298.4	.	.	.
FitRandUN	8254.1	21	8296.1	8351.1	11.9846	14	<b>0.60754</b>

**At last, the grand finale comparison: Tests of absolute fit of each side by estimator, and predicted means/variances all in ML for comparability...**

**I think random quadratic log time wins—and that’s going in chapter 6 when I revise my book!**

Tests of Absolute Model Fit Per Side		Means Side Test			Variance Side Test		
		Chi-Square	Test DF	p-value	Chi-Square	Test DF	p-value
3b	Polynomial: Random Quadratic Time	9.06	3	0.030	35.76	14	0.001
4c	Piecewise: Random Slope12, Random Slope26	4.74	3	<b>0.195</b>	15.77	14	<b>0.327</b>
6c	Negative Exponential: Random Asymptote, Random Amount, Fixed Rate	2.94	3	<b>0.404</b>	46.50	17	0.000
<b>NEW</b>	<b>Random Quadratic Log Time</b>	<b>1.31</b>	<b>3</b>	<b>0.726</b>	<b>11.98</b>	<b>14</b>	<b>0.608</b>



